
In the lab assignments for this course, we will be using the Verilog hardware description language to
model processors, memories, and networks at the register-transfer level (RTL). This tutorial briefly
reviews the basics of the Verilog hardware description language, but primarily focuses on the specific
Verilog development, testing, and evaluation framework as well as the coding conventions we will
be using in the course. We will be using two primary open source tools for the lab assignments:
Icarus Verilog (iverilog) for compiling Verilog models into simulators; and GTKWave (gtkwave) for
viewing waveforms. Both tools are installed and available on amdpool. This tutorial assumes that
students have completed the Linux and Git tutorials.

To follow along with the tutorial, access the ECE computing resources, and type the commands
without the % character. In addition to working through the commands in the tutorial, you should
also try the more open-ended tasks marked with the H symbol.

Before you begin, make sure that you have sourced the ece4750-setup script or that you have added
it to your .bashrc script, which will then source the script every time you login. Sourcing the setup
script sets up the environment required for this class.

NOTE: It should be possible to experiment with this tutorial even if you are not enrolled
in the course and/or do not have access to the ECE computing resources. All of the code
for the tutorial is located here: https://github.com/cbatten/ece4750-tut3-verilog.
You will not use the ece4750-lab-admin script, and your specific environment may be
different from what is assumed in this tutorial.

1. Verilog Modeling: Synthesizable vs. Non-Synthesizable

Verilog is a powerful language that was originally intended for building simulators of hardware as
opposed to models that could automatically be transformed into hardware (e.g., synthesized to an
FPGA or ASIC). Given this, it is very easy to write Verilog code that does not actually model any kind
of realistic hardware. Indeed, we actually need this feature to be able to write clean and productive
test harnesses, assertions, and line tracing. So students must be very diligent in actively deciding
whether or not they are writing synthesizable register-transfer-level models or non-synthesizable
code.

Students’ design work will almost exclusively use synthesizable register-transfer-level (RTL) mod-
els, while students’ verification work will likely use both synthesizable and non-synthesizable con-
structs. There is nothing wrong in using non-synthesizable constructs in a test harness since the test
harness is not meant to model hardware just test hardware. Similarly, it is acceptable to include a
limited amount of non-synthesizable code in students’ designs for the sole purpose of debugging,
assertions, and line tracing. If the student includes non-synthesizable code in their actual design (i.e.,
not the test harness), they must explicitly demarcate this code by wrapping it in ‘ifndef SYNTHESIS
and ‘endif. This explicitly documents the code as non-synthesizable and aids automated tools in
removing this code before synthesizing the design.

Appendix A includes a table that outlines which Verilog constructs are allowed in synthesizable
RTL, which constructs are allowed in synthesizable RTL with limitations, and which constructs are
explicitly not allowed in synthesizable RTL. Note that pretty much any construct is allowed when
implementing test harnesses for verification. There are also no limits on using the Verilog preproces-
sor, since the preprocessing step happens at compile time.

2. Verilog Basics: Data Types, Operators, and Conditionals

We will begin by writing some very basic code to explore Verilog data types, operators, and condi-
tionals. We will not be modeling actual hardware yet; we are just experimenting with the language.
Start by creating a directory to work in.

% source setup-ece4750.sh
% cd ${HOME}
% mkdir ece4750-tut3-verilog
% cd ece4750-tut3-verilog

2.1. Hello World

Create a new Verilog source file named hello-world.v with the contents shown in Figure 1 using
your favorite text editor. A module is the fundamental hardware building block in Verilog, but for
now we are simply using it to encapsulate an initial block. The initial block specifies code which
should be executed “at the beginning of time” when the simulator starts. Since real hardware cannot
do anything “at the beginning of time” initial blocks are not synthesizable and you should not
use them in the synthesizable portion of your designs. However, initial blocks can be useful for
test harnesses and when experimenting with the Verilog language. The initial block in Figure 1
contains a single call to the display system task which will output the given string to the console.

We will be using iverilog to compile Verilog models into simulators that we can then use to test
the model’s functionality and evaluate the model’s performance. You can run iverilog as follows
to compile hello-world.v.

% cd ${HOME}/ece4750-tut3-verilog
% iverilog -g2012 -o hello-world hello-world.v

The -g2012 option tells iverilog to accept newer SystemVerilog syntax, and the -o specifies the
name of the simulator that iverilog will create. You can run this simulator as follows.

% cd ${HOME}/ece4750-tut3-verilog
% ./hello-world

1 module top;
2 initial begin
3 $display( "Hello World!" );
4 end
5 endmodule

Code at https://github.com/cbatten/x/blob/master/ex-basics-hello-world.v

Figure 1: Verilog Basics: Display Statement – The obligatory “Hello, World!” program to compiling
a basic Verilog program.

Logical Operators

& bitwise AND
| bitwise OR
^ bitwise XOR
^~ bitwise XNOR
~ bitwise NOT

&& boolean AND
|| boolean OR
! boolean NOT

Arithmetic Operators

+ addition
- subtraction

Reduction Operators

& reduce via AND
~& reduce via NAND
| reduce via OR
~| reduce via NOR
^ reduce via XOR
^~ reduce via XNOR

Shift Operators

>> shift right
<< shift left
>>> arithmetic shift right

Relational Operators

== equal
!= not equal
> greater than
>= greater than or equals
< less than
<= less than or equals

Other Operators

{} concatenate
{N{}} replicate N times

Table 1: Table of Verilog Operators – Not all Verilog operators are shown, just those operators that
are acceptable for use in the synthesizable RTL portion of students’ designs.

As discussed in the Linux tutorial you can compile the Verilog and run the simulator in a single step.

% cd ${HOME}/ece4750-tut3-verilog
% iverilog -g2012 -o hello-world hello-world.v && ./hello-world

This makes it easy to edit the Verilog source file, quickly recompile, and test your changes by switch-
ing to your terminal, pressing the up-arrow key, and then pressing enter.

H To-Do On Your Own: Edit the string that is displayed in this simple program, recompile, and rerun
the simulator.

2.2. Logic Data Types

To understand any new modeling language we usually start by exploring the primitive data types
for representing values in a model. Verilog uses a combination of the wire and reg keywords which
interact in subtle and confusing ways. SystemVerilog has simplified modeling by introducing the
logic data type. We will be exclusively using logic as the general-purpose, hardware-centric data
type for modeling a single bit or multiple bits. Each bit can take on one of four values: 0, 1, X, Z.
X is used to represent unknown values (e.g., the state of a register on reset). Z is used to represent
high-impedance values. Although we will use variables with X values in this course, we will not use
variables with Z values (although you may see Z values if you forget to connect an input port of a
module).

Table 1 shows the operators that we will be primarily using in this course. Note that Verilog and Sys-
temVerilog support additional operators including * for multiplication, / for division, % for modulus,
** for exponent, and ===/!=== for special equality checks. There may occasionally be reasons to use
one of these operators in your test harnesses. However, these operators are either not synthesizable
or synthesize poorly, so students are not allowed to use these operators in the synthesizable portion
of their designs.

1 module top;
2

3 // Declare single -bit logic variables.
4

5 logic a;
6 logic b;
7 logic c;
8

9 initial begin
10

11 // Single -bit literals
12

13 a = 1'b0; $display( "1'b0 = %x ", a );
14 a = 1'b1; $display( "1'b1 = %x ", a );
15 a = 1'bx; $display( "1'bx = %x ", a );
16 a = 1'bz; $display( "1'bz = %x ", a );
17

18 // Bitwise logical operators for doing AND , OR , XOR , and NOT
19

20 a = 1'b0;
21 b = 1'b1;
22

23 c = a & b; $display( "0 & 1 = %x ", c );
24 c = a | b; $display( "0 | 1 = %x ", c );
25 c = a ^ b; $display( "0 ^ 1 = %x ", c );
26 c = ~b; $display( "~1 = %x ", c );
27

28 // Bitwise logical operators for doing AND , OR , XOR , and NOT with X
29

30 a = 1'b0;
31 b = 1'bx;
32

33 c = a & b; $display( "0 & x = %x ", c );
34 c = a | b; $display( "0 | x = %x ", c );
35 c = a ^ b; $display( "0 ^ x = %x ", c );
36 c = ~b; $display( "~x = %x ", c );
37

38 // Boolean logical operators
39

40 a = 1'b0;
41 b = 1'b1;
42

43 c = a && b; $display( "0 && 1 = %x ", c );
44 c = a || b; $display( "0 || 1 = %x ", c );
45 c = !b; $display( "!1 = %x ", c );
46

47 end
48

49 endmodule

Code at https://github.com/cbatten/x/blob/master/ex-basics-logic-sbit.v

Figure 2: Verilog Basics: Single-Bit Logic and Logical Operators – Experimenting with single-bit
logic variables and literals, logical bitwise operators, and logical boolean operators.

Figure 2 shows an example program that illustrates single-bit logic types. Create a new Verilog
source file named logic-sbit.v and copy some or all of this code. Compile this source file and run
the resulting simulator.

Lines 13–16 illustrate how to write single-bit literals to express constant values. Lines 23–26 illustrate
basic bitwise logical operators (&, |, ˆ, ~). Whenever we consider an expression in Verilog, we should
always ask ourselves, “What will happen if one of the inputs is an X?” Lines 33–36 illustrate what
happens if the second operand is an X for bitwise logical operators. Recall that X means “unknown”.
If we OR the value 0 with an unknown value we cannot know the result. If the unknown value is

0, then the result should be 0, but if the unknown value is 1, then the result should be 1. So Verilog
specifies that in this case the value of the expression is X. Notice what happens if we AND the value
0 with an unknown value. In this case, we can guarantee that for any value for the second operand
the result will always be 0, so Verilog specifies the value of the expression is 0.

In addition to the basic bitwise logical operators, Verilog also defines three boolean logical operators
(&&, ||, !). These operators are effectively the same as the basic logical operators (&, |, ~) when
operating on single-bit logic values. The difference is really in the designer’s intent. Using &&, ||, !
suggests that the designer is implementing a boolean logic expression as opposed to doing low-level
bit manipulation.

H To-Do On Your Own: Experiment with more complicated multi-stage logic expressions by writing
the boolean logic equations for a one-bit full-adder. Use the display system task to output the
result to the console. Experiment with using X input values as inputs to these logic expressions.

Multi-bit logic types are used for modeling bit vectors. Figure 3 shows an example program that
illustrates multi-bit logic types. Create a new Verilog source file named logic-mbit.v and copy
some or all of this code. Compile this source file and run the resulting simulator.

Lines 5–8 declares multi-bit logic variables. The square brackets contain the index of the most-
significant and the least-significant bit. In this course, you should always use zero as the index
of the least significant bit. Note that to declare a four-bit logic value, we use [3:0] not [4:0].

Lines 14–17 illustrate multi-bit literals that can be used to declare constant values. These literals have
the following general syntax: <bitwidth>’<base><number> where <base> can be b for binary, h for
hexadecimal, or d for decimal. It is legal to include underscores in the literal, which can be helpful
for improving the readability of long literals.

Lines 24–28 illustrate multi-bit versions of the basic bitwise logic operators. As before, we should
always ask ourselves, “What will happen if one of the inputs is an X?” Lines 35–39 illustrate what
happens if two bits in the second value are Xs. Note that some bits in the result are X and some can
be guaranteed to be either a 0 or 1.

Lines 45–50 illustrate the reduction operators. These operators take a multi-bit logic value and com-
bine all of the bits into a single-bit value. For example, the OR reduction operator (|) will OR all of
the bits together.

H To-Do On Your Own: We will use relational operators (e.g., ==) to compare two multi-bit logic
values, but see if you can achieve the same effect with the bitwise XOR/XNOR operators and
OR/NOR reduction operators.

1 module top;
2

3 // Declare multi -bit logic variables
4

5 logic [ 3:0] A; // 4-bit logic variable
6 logic [ 3:0] B; // 4-bit logic variable
7 logic [ 3:0] C; // 4-bit logic variable
8 logic [11:0] D; // 12-bit logic variable
9

10 initial begin
11

12 // Multi -bit literals
13

14 A = 4'b0101; $display( "4'b0101 = %x", A );
15 D = 12' b1100_1010_0101; $display( "12' b1100_1010_0101 = %x", D );
16 D = 12'hca5; $display( "12'hca5 = %x", D );
17 D = 12'd1058; $display( "12'd1058 = %x", D );
18

19 // Bitwise logical operators for doing AND , OR, XOR , and NOT
20

21 A = 4'b0101;
22 B = 4'b0011;
23

24 C = A & B; $display( "4'b0101 & 4'b0011 = %b", C );
25 C = A | B; $display( "4'b0101 | 4'b0011 = %b", C );
26 C = A ^ B; $display( "4'b0101 ^ 4'b0011 = %b", C );
27 C = A ^~ B; $display( "4'b0101 ^~ 4'b0011 = %b", C );
28 C = ~B; $display( "~4'b0011 = %b", C );
29

30 // Bitwise logical operators when some bits are X
31

32 A = 4'b0101;
33 B = 4'b00xx;
34

35 C = A & B; $display( "4'b0101 & 4'b00xx = %b", C );
36 C = A | B; $display( "4'b0101 | 4'b00xx = %b", C );
37 C = A ^ B; $display( "4'b0101 ^ 4'b00xx = %b", C );
38 C = A ^~ B; $display( "4'b0101 ^~ 4'b00xx = %b", C );
39 C = ~B; $display( "~4'b00xx = %b", C );
40

41 // Reduction operators
42

43 A = 4'b0101;
44

45 C = &A; $display( " & 4'b0101 = %b", C );
46 C = ~&A; $display( "~& 4'b0101 = %b", C );
47 C = |A; $display( " | 4'b0101 = %b", C );
48 C = ~|A; $display( "~| 4'b0101 = %b", C );
49 C = ^A; $display( "^ 4'b0101 = %b", C );
50 C = ^~A; $display( "^~ 4'b0101 = %b", C );
51

52 end
53

54 endmodule

Code at https://github.com/cbatten/x/blob/master/ex-basics-logic-mbit.v

Figure 3: Verilog Basics: Multi-Bit Logic and Logical Operators – Experimenting with multi-bit
logic variables and literals, bitwise logical operators, and reduction operators.

2.3. Shift Operators

Figure 4 illustrates three shift operators on multi-bit logic values. Create a new Verilog source file
named logic-shift.v and copy some or all of this code. Compile this source file and run the result-
ing simulator.

Notice how the logical shift operators (<<, >>) always shift in zeros, but the arithmetic right shift
operator (>>>) replicates the most-significant bit. Verilog requires that the left-hand operand to the
arithmetic shift operator be explicitly denoted as signed, which we have done using the signed
system task. We recommend this approach and avoiding the use of signed data types.

H To-Do On Your Own: Experiment different multi-bit logic values and shift amounts.

1 module top;
2

3 logic [7:0] A;
4 logic [7:0] B;
5 logic [7:0] C;
6

7 initial begin
8

9 // Fixed shift amount for logical shifts
10

11 A = 8'b1110_0101;
12

13 C = A << 1; $display( "8'b1110_0101 << 1 = %b", C );
14 C = A << 2; $display( "8'b1110_0101 << 2 = %b", C );
15 C = A << 3; $display( "8'b1110_0101 << 3 = %b", C );
16

17 C = A >> 1; $display( "8'b1110_0101 >> 1 = %b", C );
18 C = A >> 2; $display( "8'b1110_0101 >> 2 = %b", C );
19 C = A >> 3; $display( "8'b1110_0101 >> 3 = %b", C );
20

21 // Fixed shift amount for arithmetic shifts
22

23 A = 8'b0110_0100;
24

25 C = $signed(A) >>> 1; $display( "8'b0110_0100 >>> 1 = %b", C );
26 C = $signed(A) >>> 2; $display( "8'b0110_0100 >>> 2 = %b", C );
27 C = $signed(A) >>> 3; $display( "8'b0110_0100 >>> 3 = %b", C );
28

29 A = 8'b1110_0101;
30

31 C = $signed(A) >>> 1; $display( "8'b1110_0101 >>> 1 = %b", C );
32 C = $signed(A) >>> 2; $display( "8'b1110_0101 >>> 2 = %b", C );
33 C = $signed(A) >>> 3; $display( "8'b1110_0101 >>> 3 = %b", C );
34

35 // Variable shift amount for logical shifts
36

37 A = 8'b1110_0101;
38 B = 8'd2;
39

40 C = A << B; $display( "8'b1110_0101 << 2 = %b", C );
41 C = A >> B; $display( "8'b1110_0101 >> 2 = %b", C );
42

43 end
44

45 endmodule

Code at https://github.com/cbatten/x/blob/master/ex-basics-logic-shift.v

Figure 4: Verilog Basics: Shift Operators – Experimenting with logical and arithmetic shift operators
and fixed/variable shift amounts.

2.4. Arithmetic Operators

Figure 5 illustrates the addition and subtraction operators for multi-bit logic values. Create a new
Verilog source file named logic-arith.v and copy some or all of this code. Compile this source file
and run the resulting simulator.

These operators treat the multi-bit logic values as unsigned integers. Although Verilog does include
support for signed arithmetic, these constructs may not be synthesizable so you are required to use
only unsigned arithmetic. Also recall that *, /, %, ** are not allowed in the synthesizable portion of
your design.

Note that carefully considering the bitwidths of the input and output variables is important. Lines 22–
23 illustrate overflow and underflow. You can see that if you overflow the bitwidth of the output
variable then the result will simply wrap around. Similarly, since we are using unsigned arithmetic
negative numbers wrap around. This is also called modular arithmetic.

H To-Do On Your Own: Try writing some code which does a sequence of additions resulting in
overflow and then a sequence of subtractions that essentially undo the overflow. For example,
try 200 + 400 + 400 - 400 - 400. Does this expression produce the expected answer even
though the intermediate values overflowed?

1 module top;
2

3 logic [7:0] A;
4 logic [7:0] B;
5 logic [7:0] C;
6

7 initial begin
8

9 // Basic arithmetic with no overflow or underflow
10

11 A = 8'd28;
12 B = 8'd15;
13

14 C = A + B; $display( "8'd28 + 8'd15 = %d", C );
15 C = A - B; $display( "8'd28 + 8'd15 = %d", C );
16

17 // Basic arithmetic with overflow and underflow
18

19 A = 8'd250;
20 B = 8'd15;
21

22 C = A + B; $display( "8'd250 + 8'd15 = %d", C );
23 C = B - A; $display( "8'd15 - 8'd250 = %d", C );
24

25 end
26

27 endmodule

Code at https://github.com/cbatten/x/blob/master/ex-basics-logic-arith.v

Figure 5: Verilog Basics: Arithmetic Operators – Experimenting with arithmetic operators for addi-
tion and subtraction.

2.5. Relational Operators

Figure 6 illustrates the relational operators used for comparing two multi-bit logic values. Create a
new Verilog source file named logic-relop.v and copy some or all of this code. Compile this source
file and run the resulting simulator.

Lines 28–33 illustrate what happens if some of the bits are Xs for these relational operators. Notice
that we can still determine two values are not equal even if some bits are unknown. If the bits we do
know are different then the unknown bits do not matter; we can guarantee that the full bit vectors
are not equal. So in this example, since we know that the top-two bits in 4’b1100 and 4’b10xx then
we can guarantee that the two values are not equal even though the bottom two bits of one operand
are unknown.

The <, >, <=, >= operators behave slightly differently than the == and != operators when considering
values with Xs. In this example, we should be able to guarantee that 4’b1100 is always greater than
4’b10xx (assuming these are unsigned values), since no matter what the bottom two bits are in the
second operand it cannot be greater than the first operand. However, if you run this simulation, then
you will see that the result is still X. This is not a bug and is correct according to the Verilog language
specification. This is a great example of how Verilog has relatively complicated and sometimes in-
consistent language semantics. Originally, there was no Verilog standard. The most common Verilog
simulator was the de-factor language standard. I imagine the reason there is this difference between
how == and < handle X values is simply because in the very first Verilog simulators it was the most
efficient solution. These kind of “simulator implementation issues” can be found throughout the
Verilog standard.

Lines 40–43 illustrates signed comparisons using the signed system task to to interpret the unsigned
input operands as signed values. To simplify our designs, we do not allow students to use signed
types. We should explicitly use the signed system task whenever we need to ensure signed compar-
isons.

H To-Do On Your Own: Try composing relational operators with the boolean logic operators we
learned about earlier in this section to create more complicated expressions.

1 module top;
2

3 // Declare multi -bit logic variables
4

5 logic a; // 1-bit logic variable
6 logic [ 3:0] A; // 4-bit logic variable
7 logic [ 3:0] B; // 4-bit logic variable
8

9 initial begin
10

11 // Relational operators
12

13 A = 4'd15; B = 4'd09;
14

15 a = ( A == B ); $display( "(15 == 9) = %x", a );
16 a = ( A != B ); $display( "(15 != 9) = %x", a );
17 a = ( A > B ); $display( "(15 > 9) = %x", a );
18 a = ( A >= B ); $display( "(15 >= 9) = %x", a );
19 a = ( A < B ); $display( "(15 < 9) = %x", a );
20 a = ( A <= B ); $display( "(15 <= 9) = %x", a );
21

22 // Relational operators when some bits are X
23

24 A = 4'b1100; B = 4'b10xx;
25

26 a = ( A == B ); $display( "(4'b1100 == 4'b10xx) = %x", a );
27 a = ( A != B ); $display( "(4'b1100 != 4'b10xx) = %x", a );
28 a = ( A > B ); $display( "(4'b1100 > 4'b10xx) = %x", a );
29 a = ( A < B ); $display( "(4'b1100 < 4'b10xx) = %x", a );
30

31 // Signed relational operators
32

33 A = 4'b1111; // -1 in twos complement
34 B = 4'd0001; // 1 in twos complement
35

36 a = ( A > B ); $display( "(-1 > 1) = %x", a );
37 a = ( A < B ); $display( "(-1 < 1) = %x", a );
38 a = ( $signed(A) > $signed(B) ); $display( "(-1 > 1) = %x", a );
39 a = ( $signed(A) < $signed(B) ); $display( "(-1 < 1) = %x", a );
40

41 end
42

43 endmodule

Code at https://github.com/cbatten/x/blob/master/ex-basics-logic-relop.v

Figure 6: Verilog Basics: Relational Operators – Experimenting with relational operators.

2.6. Concatenation Operators

Figure 7 illustrates the concatenation operators used for creating larger bit vectors from multiple
smaller bit vectors. Create a new Verilog source file named logic-concat.v and copy some or all of
this code. Compile this source file and run the resulting simulator.

Lines 18–20 illustrate concatenating three four-bit logic variables and then assigning the result to a
12-bit logic variable. Lines 25–26 illustrate concatenating a four-bit logic variable with an eight-bit
logic variable. The repeat operator can be used to duplicate a given logic variable multiple times
when creating larger bit vectors. On Line 33, we replicate a four-bit logic value three times to create
a 12-bit logic value.

H To-Do On Your Own: Experiment with different variations of concatenation and the repeat opera-
tor to create interesting bit patterns.

1 module top;
2

3 logic [ 3:0] A; // 4-bit logic variable
4 logic [ 3:0] B; // 4-bit logic variable
5 logic [ 3:0] C; // 4-bit logic variable
6 logic [ 7:0] D; // 18-bit logic variable
7 logic [11:0] E; // 12-bit logic variable
8

9 initial begin
10

11 // Basic concatenation
12

13 A = 4'ha;
14 B = 4'hb;
15 C = 4'hc;
16 D = 8'hde;
17

18 E = { A, B, C }; $display( "{ 4'ha, 4'hb , 4'hc } = %x", E );
19 E = { C, A, B }; $display( "{ 4'hc, 4'ha , 4'hb } = %x", E );
20 E = { B, C, A }; $display( "{ 4'hb, 4'hc , 4'ha } = %x", E );
21

22 E = { A, D }; $display( "{ 4'ha, 8'hde } = %x", E );
23 E = { D, A }; $display( "{ 8'hde , 4'ha } = %x", E );
24

25 E = { A, 8'hf0 }; $display( "{ 4'ha, 8'hf0 } = %x", E );
26 E = { 8'hf0 , A }; $display( "{ 8'hf0 , 4'ha } = %x", E );
27

28 // Repeat operator
29

30 A = 4'ha;
31 B = 4'hb;
32

33 E = { 3{A} }; $display( "{ 4'ha, 4'ha , 4'ha } = %x", E );
34 E = { A, {2{B}} }; $display( "{ 4'ha, 4'hb , 4'hb } = %x", E );
35

36 end
37

38 endmodule

Code at https://github.com/cbatten/x/blob/master/ex-basics-logic-concat.v

Figure 7: Verilog Basics: Concatenation Operators – Experimenting with the basic concatenation
operator and the repeat operator.

2.7. Enum Data Types

The logic data type will be the most common data type we use in our synthesizable RTL since a
logic variable has a direct one-to-one correspondence with a bit vector in hardware. However, there
are certain cases where using a logic variable can be quite tedious and error prone. SystemVerilog
has introduced two new kinds of user-defined types that can greatly simplify some portions of our
designs. In this subsection, we introduce the enum type which enables declaring variables that can
only take on a predefined list of labels.

Figure 8 illustrates creating and using an enum type for holding a state variable which can take on
one of four labels. Create a new Verilog source file named enum.v and copy all of this code. Compile
this source file and run the resulting simulator.

Lines 3–8 declare a new enum type named state_t. Note that state_t is not a new variable but is
instead a new type. We will use the _t suffix to distinguish type names from variable names. Note
that after the enum keyword we have included a base type of logic [$clog2(4)-1:0]. This base type
specifies how we wish variables of this new type to be stored. In this case, we are specifying that
state_t variables should be encoded as a two-bit logic value. The clog2 system task calculates the
number of bits in the given argument; it is very useful when writing more parameterized code. So
in this situation we just need to pass in the number of labels in the enum to clog2. SystemVerilog
actually provides many different ways to create enum types including anonymous types, types where
we do not specify the base type, or types where we explicitly define the value for each label. In this
course, you should limit yourself to the exact syntax shown in this example.

Line 14 declares a new variable of type state_t. This is the first time we have seen a variable which
has a type other than logic. The ability to introduce new user-defined types is one of the more
powerful features of SystemVerilog. Lines 21–24 sets the state variable using the labels declared as
part of the new state_t type. Lines 28–40 compare the value of the state variable with these same
labels, and these comparisons can be used to take different action based on the current value.

There are several advantages to using an enum type compared to the basic logic type to represent
a variable that can hold one of several labels including: (1) more directly capturing the designer’s
intent to improve code quality; (2) preventing mistakes by eliminating the possibility of defining
labels with the same value or defining label values that are too large to fit in the underlying storage;
and (3) preventing mistakes when assigning variables of a different type to an enum variable.

H To-Do On Your Own: Create your own new enum type for the state variable we will use in the GCD
example later in this tutorial. The new enum type should be called state_t and it should support
three different labels: STATE_IDLE, STATE_CALC, STATE_DONE. Write some code to set and compare
the value of a corresponding state variable.

1 // Declare enum type
2

3 typedef enum logic [$clog2 (4) -1:0] {
4 STATE_A ,
5 STATE_B ,
6 STATE_C ,
7 STATE_D
8 } state_t;
9

10 module top;
11

12 // Declare variables
13

14 state_t state;
15 logic result;
16

17 initial begin
18

19 // Enum lable literals
20

21 state = STATE_A; $display( "STATE_A = %d", state );
22 state = STATE_B; $display( "STATE_B = %d", state );
23 state = STATE_C; $display( "STATE_C = %d", state );
24 state = STATE_D; $display( "STATE_D = %d", state );
25

26 // Comparisons
27

28 state = STATE_A;
29

30 result = ( state == STATE_A );
31 $display( "( STATE_A == STATE_A ) = %x", result );
32

33 result = ( state == STATE_B );
34 $display( "( STATE_A == STATE_B ) = %x", result );
35

36 result = ( state != STATE_A );
37 $display( "( STATE_A != STATE_A ) = %x", result );
38

39 result = ( state != STATE_B );
40 $display( "( STATE_A != STATE_B ) = %x", result );
41

42 end
43

44 endmodule

Code at https://github.com/cbatten/x/blob/master/ex-basics-enum.v

Figure 8: Verilog Basics: Enum Data Types – Experimenting with enum data types including setting
the value of an enum using a label and using the equality operator.

2.8. Struct Data Types

Figure 9 illustrates creating and using a struct type for holding a variable with predefined named
bit fields. Create a new Verilog source file named struct.v and copy all of this code. Compile this
source file and run the resulting simulator.

Lines 3–7 declare a new struct type named point_t. Again note that point_t is not a new variable
but is instead a new type. As before we use the _t suffix to distinguish type names from variable
names. Note that after the struct keyword we have included the packed keyword which specifies
that variables of this type should have an equivalent underlying logic storage. SystemVerilog also
includes support for unpacked structs, but in this course, you should limit yourself to the exact
syntax shown in this example. In addition to declaring the name of the new struct type, we also
declare the named bit fields within the new struct type. The order of these bit fields is important;
the first field will go in the most significant position of the underlying logic storage, the second field
will go in the next position, and so on.

Lines 13–14 declare two new variables of type point_t. Line 18 declares a new logic variable with a
bitwidth large enough to hold a variable of type point_t. We can use the bits system task to easily
determine the total number of bits required to store a struct type. Lines 24–26 set the three fields of
the point variable and Lines 28–30 read these three fields in order to display them. Line 34 copies
one point variable into another point variable. Line 42 and 49 illustrate how to convert a point
variable to/from a basic logic variable.

There are several advantages to using a struct type compared to the basic logic type to represent
a variable with a predefined set of named bit fields including: (1) more directly capturing the de-
signer’s intent to improve code quality; (2) reducing the syntactic overhead of managing bit fields;
and (3) preventing mistakes in modifying bit fields and in accessing bit fields.

H To-Do On Your Own: Create your own new struct type for holding the request messages we
will use in the GCD example later in this tutorial. The new struct type should be called
gcd_req_msg_t and it should include two fields: a 32-bit field named a and a 32-bit field named
b. Write some code to write and read these fields.

1 // Declare struct type
2

3 typedef struct packed { // Packed format:
4 logic [3:0] x; // 11 8 7 4 3 0
5 logic [3:0] y; // +----+----+----+
6 logic [3:0] z; // | x | y | z |
7 } point_t; // +----+----+----+
8

9 module top;
10

11 // Declare variables
12

13 point_t point_a;
14 point_t point_b;
15

16 // Declare other variables using $bits()
17

18 logic [$bits(point_t ) -1:0] point_bits;
19

20 initial begin
21

22 // Reading and writing fields
23

24 point_a.x = 4'h3;
25 point_a.y = 4'h4;
26 point_a.z = 4'h5;
27

28 $display( "point_a.x = %x", point_a.x );
29 $display( "point_a.y = %x", point_a.y );
30 $display( "point_a.z = %x", point_a.z );
31

32 // Assign structs
33

34 point_b = point_a;
35

36 $display( "point_b.x = %x", point_b.x );
37 $display( "point_b.y = %x", point_b.y );
38 $display( "point_b.z = %x", point_b.z );
39

40 // Assign structs to bit vector
41

42 point_bits = point_a;
43

44 $display( "point_bits = %x", point_bits );
45

46 // Assign bit vector to struct
47

48 point_bits = { 4'd13 , 4'd9 , 4'd3 };
49 point_a = point_bits;
50

51 $display( "point_a.x = %x", point_a.x );
52 $display( "point_a.y = %x", point_a.y );
53 $display( "point_a.z = %x", point_a.z );
54

55 end
56

57 endmodule

Code at https://github.com/cbatten/x/blob/master/ex-basics-struct.v

Figure 9: Verilog Basics: Struct Data Types – Experimenting with struct data types including
read/writing fields and converting to/from logic bit vectors.

2.9. Ternary Operator

Figure 10 illustrates using the ternary operator for
conditional execution. Create a new Verilog source
file named ternary.v and copy some or all of this
code. Compile this source file and run the resulting
simulator.

Lines 12–19 illustrate using the ternary operator to
choose what value to assign to the logic variable
c. We can nest multiple ternary operators to com-
pactly create expressions with multiple conditions.
Lines 23–31 illustrate using four levels of nesting to
choose among four different values for assigning c.

Lines 35–53 illustrate how the ternary operator
functions if the conditional is unknown. In
Lines 35–43, all bits of the conditional are un-
known, while in Lines 45–53 only one bit of the
conditional is unknown. If you examine the out-
put from this simulator, you will see that Verilog
semantics require any bits which can be guaranteed
to be either 0 or 1 to be set as such, while the re-
maining bits are set to X. Regardless of the condi-
tion, the upper five bits of c are guaranteed to be
00001.

Note that the four ternary operators cover all pos-
sible combinations of the two-bit input, so the fi-
nal value (i.e., 8’h0e) will never be used. In other
words, if the conditionals contain unknowns this
does not mean the condition evaluates to false. This
is very different from the if statements described
in the next subsection.

Aside: For some reason, many students insist on
writing code like this:

a = ( cond_a ) ? 1’b1 : 1’b0;
b = ( cond_b ) ? 1’b0 : 1’b1;

This obfuscates the code and is not necessary. We
are using a ternary operator to simply choose be-
tween 0 or 1. You should just assign the result of
the conditional expression to a and b like this:

a = ( cond_a );
b = !( cond_b );

H To-Do On Your Own: Experiment with different
uses of the ternary operator.

1 module top;
2

3 logic [7:0] a;
4 logic [7:0] b;
5 logic [7:0] c;
6 logic [1:0] sel;
7

8 initial begin
9

10 // ternary statement
11

12 a = 8'd30;
13 b = 8'd16;
14

15 c = ( a < b ) ? 15 : 14;
16 $display( "c = %d", c );
17

18 c = ( b < a ) ? 15 : 14;
19 $display( "c = %d", c );
20

21 // nested ternary statement
22

23 sel = 2'b01;
24

25 c = ( sel == 2'b00 ) ? 8'h0a
26 : ( sel == 2'b01 ) ? 8'h0b
27 : ( sel == 2'b10 ) ? 8'h0c
28 : ( sel == 2'b11 ) ? 8'h0d
29 : 8'h0e;
30

31 $display( "sel = 1, c = %b", c );
32

33 // nested ternary statement w/ X
34

35 sel = 2'bxx;
36

37 c = ( sel == 2'b00 ) ? 8'h0a
38 : ( sel == 2'b01 ) ? 8'h0b
39 : ( sel == 2'b10 ) ? 8'h0c
40 : ( sel == 2'b11 ) ? 8'h0d
41 : 8'h0e;
42

43 $display( "sel = x, c = %b", c );
44

45 sel = 2'bx0;
46

47 c = ( sel == 2'b00 ) ? 8'h0a
48 : ( sel == 2'b01 ) ? 8'h0b
49 : ( sel == 2'b10 ) ? 8'h0c
50 : ( sel == 2'b11 ) ? 8'h0d
51 : 8'h0e;
52

53 $display( "sel = x, c = %b", c );
54

55 end
56

57 endmodule

Code at https://github.com/cbatten/x/blob/
master/ex-basics-ternary.v

Figure 10: Verilog Basics: Ternary Operator
– Experimenting with the ternary operator in-
cluding nested statements and what happens
if the conditional includes an unknown.

2.10. If Statements

Figure 11 illustrates using if statements. Create a
new Verilog source file named if.v and copy some
or all of this code. Compile this source file and run
the resulting simulator.

The if statement resembles similar constructs in
many other programming languages. Lines 11–20
illustrate basic if statements and Lines 24–33 illus-
trate if/else statements.

There are some subtle issues involved in how an
if statement handles X values in the conditional.
Lines 37–46 illustrate this issue. The sel value in
this example is a single-bit X. What would we ex-
pect the value of a to be after this if statement?
Since the conditional is unknown, we might expect
any variables that are written from within the if
statement to also be unknown. In other words, we
might expect the value of a to be X after this if
statement. If you run this example code, you will
see that the value of a is actually 8’h0b. This means
that an X value in the conditional for an if state-
ment is not treated as unknown but is instead sim-
ply treated as if the conditional evaluated to false!
This issue is called X optimism since unknowns are
essentially optimistically turned into known val-
ues. X optimism can cause subtle simulation/syn-
thesis mismatch issues. If you are interested, there
are several resources on the public course webpage
that go into much more detail. For this course, we
don’t need to worry too much about X optimism
since we are not actually synthesizing our designs.

H To-Do On Your Own: Experiment with different
if statements including nested if statements.
Experiment with what happens when the con-
ditional is unknown.

1 module top;
2

3 logic [7:0] a;
4 logic [7:0] b;
5 logic sel;
6

7 initial begin
8

9 // if statement
10

11 a = 8'd30;
12 b = 8'd16;
13

14 if ( a == b ) begin
15 $display( "30 == 16" );
16 end
17

18 if ( a != b ) begin
19 $display( "30 != 16" );
20 end
21

22 // if else statement
23

24 sel = 1'b1;
25

26 if ( sel == 1'b0 ) begin
27 a = 8'h0a;
28 end
29 else begin
30 a = 8'h0b;
31 end
32

33 $display( "sel = 1, a = %x ", a );
34

35 // if else statement w/ X
36

37 sel = 1'bx;
38

39 if ( sel == 1'b0 ) begin
40 a = 8'h0a;
41 end
42 else begin
43 a = 8'h0b;
44 end
45

46 $display( "sel = x, a = %x ", a );
47

48 end
49

50 endmodule

Code at https://github.com/cbatten/x/blob/
master/ex-basics-if.v

Figure 11: Verilog Basics: If Statements –
Experimenting with if statements including
what happens if the conditional includes an
unknown.

2.11. Case Statements

Figure 12 illustrates using case statements. Create
a new Verilog source file named case.v and copy
some or all of this code. Compile this source file
and run the resulting simulator.

The case statement resembles similar constructs in
many other programming languages. Lines 12–
22 illustrate a basic case statement where a two-
bit sel variable is used to choose one of four case
items.

There are similar issues as with the if statement in
terms of how case statements handle X values in
the conditional. In Lines 26–36, the sel variable is
set to all Xs. We might expect since the input to
the case statement is unknown the output should
also be unknown. However, if we look at the value
of a after executing this case statement it will be
8’h0e. In other words, if there is an X in the in-
put to the case statement, then the case statement
will fall through to the default case. In order to
avoid X optimism, we recommend students always
include a default case that sets all of the output
variables to Xs.

Notice that it is valid syntax to use X values in the
case items, as shown on Lines 48–49. These will
actually match Xs in the input condition, which is
almost certainly not what you want. This does not
model any kind of real hardware; we cannot check
for Xs in hardware since in real hardware an un-
known must be known (i.e., all Xs will either be a
0 or a 1 in real hardware). Given this, you should
never use Xs in the case items for a case statement.

H To-Do On Your Own: Experiment with a larger
case statement for a sel variable with three in-
stead of two bits.

1 module top;
2

3 // Declaring Variables
4

5 logic [1:0] sel;
6 logic [7:0] a;
7

8 initial begin
9

10 // case statement
11

12 sel = 2'b01;
13

14 case ( sel )
15 2'b00 : a = 8'h0a;
16 2'b01 : a = 8'h0b;
17 2'b10 : a = 8'h0c;
18 2'b11 : a = 8'h0d;
19 default : a = 8'h0e;
20 endcase
21

22 $display( "sel = 01, a = %x", a );
23

24 // case statement w/ X
25

26 sel = 2'bxx;
27

28 case ( sel )
29 2'b00 : a = 8'h0a;
30 2'b01 : a = 8'h0b;
31 2'b10 : a = 8'h0c;
32 2'b11 : a = 8'h0d;
33 default : a = 8'h0e;
34 endcase
35

36 $display( "sel = xx , a = %x", a );
37

38 // Do not use x's in the case
39 // selection items
40

41 sel = 2'bx0;
42

43 case ( sel )
44 2'b00 : a = 8'h0a;
45 2'b01 : a = 8'h0b;
46 2'b10 : a = 8'h0c;
47 2'b11 : a = 8'h0d;
48 2'bx0 : a = 8'h0e;
49 2'bxx : a = 8'h0f;
50 default : a = 8'h00;
51 endcase
52

53 $display( "sel = x0 , a = %x", a );
54

55 end
56

57 endmodule

Code at https://github.com/cbatten/x/blob/
master/ex-basics-case.v

Figure 12: Verilog Basics: Case Statements –
Experimenting with case statements includ-
ing what happens if the selection expression
and/or the case expressions includes an un-
known.

2.12. Casez Statements

Figure 13 illustrates using casez statements. Create
a new Verilog source file named casez.v and copy
some or all of this code. Compile this source file
and run the resulting simulator.

The casez statement is very different from what
you might find in other programming languages.
The casez statement is a powerful construct that
can enable very concise hardware models, but must
be used carefully. A casez statement enables a de-
signer to do “wildcard” matching on the input vari-
able. Lines 10–23 illustrate using a casez state-
ment to implement a “leading-one detector”. This
kind of logic outputs the bit position of the least-
significant one in the input variable. We can use
? characters in the case items as wildcards that
will match either a 0 or 1 in the input variable. So
both 4’b0100 and 4’b1100 will match the fourth
case item. Implementing similar functionality us-
ing a case statement would require 16 items. Be-
sides being more verbose, using a case statement
also opens up additional opportunities for errors.

A casez statement behaves similarly to a case
statement when there are Xs in the input. Lines 27–
40 illustrate a situation where two of the bits in the
input variable are unknown. This will match the
default case and the output will be Xs.

Aside: Verilog includes a casex statement which
you should never use. The reasoning is rather sub-
tle, but to be safe stick to using casez statement if
you need wildcard matching (and only if you need
wildcard matching).

H To-Do On Your Own: Experiment with a larger
casez statement to implement a leading-one
detector for an input variable with eight instead
of four bits. How many case items would we
need if we used a case statement to implement
the same functionality?

1 module top;
2

3 logic [3:0] a;
4 logic [7:0] b;
5

6 initial begin
7

8 // casez statement
9

10 a = 4'b0100;
11

12 casez ( a )
13

14 4'b0000 : b = 8'd0;
15 4'b???1 : b = 8'd1;
16 4'b??10 : b = 8'd2;
17 4'b?100 : b = 8'd3;
18 4'b1000 : b = 8'd4;
19

20 default : b = 8'hxx;
21 endcase
22

23 $display( "a = 4'b0100 , b = %x", b );
24

25 // casez statement w/ Xs
26

27 a = 4'b01xx;
28

29 casez ( a )
30

31 4'b0000 : b = 8'd0;
32 4'b???1 : b = 8'd1;
33 4'b??10 : b = 8'd2;
34 4'b?100 : b = 8'd3;
35 4'b1000 : b = 8'd4;
36

37 default : b = 8'hxx;
38 endcase
39

40 $display( "a = 4'b01xx , b = %x", b );
41

42 end
43

44 endmodule

Code at https://github.com/cbatten/x/blob/
master/ex-basics-casez.v

Figure 13: Verilog Basics: Casez Statements –
Experimenting with casez statements to illus-
trate their use as priority selectors with wild-
cards.

3. Introduction to the Modular Verilog Build System

In the previous section, all Verilog source files were in a single directory, we manually ran iverilog
to build each small test, and the directory contained a mix of source files and compiled simulators.
While this is fine for very small projects, by the final lab we will have hundreds of Verilog source
files. Larger, more complicated projects require some kind of more structured “build system” to
organize Verilog source files, build simulators, and run these simulators for testing or evaluation.
In this course, we will be using a custom modular Verilog build system developed by the course
instructors. Use the following commands to checkout the example project we will be working on in
this tutorial:

% source setup-ece4750.sh
% ece4750-lab-admin start ece4750-tut3-verilog
% cd ${HOME}/ece4750/ece4750-tut3-verilog
% TUTROOT=${PWD}

The modular Verilog build system is based on the concept of a project containing a collection of
subprojects. The subprojects are meant to be modular in the sense that they should be relatively self-
contained and should explicitly indicate any dependencies on other subprojects. Take a few minutes
to look at the files that are included in this project:

• COPYING – Open-source license
• Makefile.in – Input used to create makefile for project
• aclocal.m4 – Autoconf fragment used by configure.ac
• configure.ac – Input used to create configure script for project
• configure – Configure script used to configure the project
• subprojs.mk – List of subprojects that make up this project
• scripts – Scripts used by the build system
• vc – Subproject containing common Verilog components
• ex-basics – Subproject containing basic code from previous section
• ex-sorter – Subproject containing example sorting unit
• ex-gcd – Subproject containing example greatest-common-divisor unit

For the most part, you will not need to modify nor really understand any of the files at the top-level
of the project; the only exception is the subprojs.mk makefile fragment that you will use to explicitly
specify which subprojects make up the project.

The basic process for building and testing your project should be familiar to anyone that has built
open-source software from source. You will need to create a build directory, configure the project,
build the project, and then run the tests. We do this with the following set of commands:

% cd ${TUTROOT}
% mkdir build
% cd build
% ../configure
% make
% make check

The configure script will check various aspects of the system you are working on to make sure it is
setup correctly. For example, the configure script will make sure that your system has an acceptable
Verilog compiler installed. When the configure script is finished it will create a new makefile in the
build directory (based on the Makefile.in file located at the top of the project) which you can use

to build, test, and evaluate your project. The make command uses the makefile to figure out how to
“make” various aspects of the project. By default, the make command will build all of the Verilog
simulators we use for evaluation. We can pass various “targets” to more specifically tell Make what
to do; for example, the check target tells Make to run all of the unit tests. You should see no failures
for the tests in the vc subproject, and you will see failures for the ex-sorter subproject that we will
fix later. We will be using make to drive our hardware development toolflow.

3.1. Creating a New Subproject

As mentioned above, each subproject should be relatively self-contained and encapsulate one rela-
tively large part of the design. For example, each of the lab assignments will be a separate subproject,
and in the final lab assignment we will leverage all of these subprojects to create a new subproject
that contains the full multicore system. Normally, the lab harness you receive will have a subproject
already created for you to use when working on the baseline and alternative design. Any extensions
you develop should be in a separate subproject to make it easier for the course staff to evaluate your
baseline and alternative designs.

We will now create a new subproject for use in the next section of the tutorial. A subproject is really
just a set of policies, naming conventions, and a few special files. Each subproject needs a name
which will be used as a prefix in many different ways. Subproject names should be short, lower-case,
and only contain letters, numbers, and dashes (-). Our new subproject will be named ex-regincr (ex
stands for PARC Example). We will use the subproject name (or a variation of the subproject name)
as a prefix for subproject filenames (e.g., ex-regincr-), Verilog macro names (e.g., EX_REGINCR_),
Verilog module names (e.g., ex_regincr_), and Make variable names (e.g., ex_regincr_). To reiter-
ate, every subproject filename, Verilog macro name, Verilog module name, and Make variable name must use
the subproject name as a prefix. This is essential to avoid namespace collisions with other subprojects.

Each subproject has its own dedicated subdirectory named after the subproject. Create a subdirectory
for our new subproject as follows:

% cd ${TUTROOT}
% mkdir ex-regincr

The key way that the build system knows about the dependencies and files of the subproject is
through a special makefile fragment which goes in the subproject’s subdirectory. The special makefile
fragment should always have the same name as the subproject with a .mk filename extension. This
makefile fragment should have at least the following four Make variables (where subproj is the
name of the subproject with dashes replaced by underscores).

• subproj_deps – List of other subprojects this subproject depends on
• subproj_srcs – List of Verilog source files with .v filename extension
• subproj_test_srcs – List of Verilog unit test source files with .t.v filename extension
• subproj_sim_srcs – List of Verilog simulator source files with .v filename extension

Let’s go ahead and create a makefile fragment for our new subproject, but we will leave all of the
variables empty for now.

% cd ${TUTROOT}/ex-regincr
% echo "
ex_regincr_deps =
ex_regincr_srcs =
ex_regincr_test_srcs =
ex_regincr_sim_srcs =

" > ex-regincr.mk

The final step in creating a new subproject to update the build system’s master list of subpro-
jects. This master list is contained in the subprojs.mk file at the top of the project. If you look in
subprojs.mk you will see that it currently lists the vc, ex-sorter, and ex-gcd subprojects. Let’s add
our new subproject to this list:

% cd ${TUTROOT}
% echo "subprojs = vc ex-basics ex-regincr ex-sorter" > subprojs.mk

We can now go ahead and try rebuilding the project:

% cd ${TUTROOT}/build
% make
% make echo-subprojs

If you added your project correctly, then this should list all of the subprojects including the new
ex-regincr subproject. Since we haven’t actually added any new Verilog source files the build sys-
tem won’t really do anything different; but we are now ready to add our first Verilog hardware
module in the next section of the tutorial.

3.2. Primary Makefile Targets

A few other makefile targets included as part of the modular Verilog build system are listed below.

• all – build all evaluation simulators in the project
• check – build and run all unit tests in the project
• clean – clean all generated content in the project
• sproj-all – build all evaluation simulators for the subproject named sproj
• sproj-check – run all unit tests for the subproject named sproj
• sproj-eval – evaluate all simulators for the subproject named sproj
• sproj-clean – clean all generated content for the subproject named sproj

For example, the following commands will run all the unit tests for just the vc subproject and then
clean all generated content for the vc subproject.

% cd ${TUTROOT}/build
% make vc-check
% make vc-clean

4. Registered Incrementer: Behavioral Verilog Development and Verification

In this section, we will create our very first Verilog hardware model and learn how to test this module
using waveforms, ad-hoc testing, and a simple unit testing framework. It is good design practice to
usually draw some kind of diagram of the hardware we wish to model before starting to develop the
corresponding Verilog model. This diagram might be a block-level diagram, a datapath diagram, a
finite-state-machine diagram, or even a control signal table; the more we can structure our Verilog
code to match this diagram the more confident we can be that our model actually models what
we think it does. In this section, we wish to model the eight-bit registered incrementer shown in
Figure 14.

in
8b 8b

out+1 +1

Figure 14: Block Diagram for Registered Incre-
menter – An eight-bit registered +2 incrementer with
an eight-bit input port, an eight-bit output port, and
an (implicit) clock input.

1 // ========================================================================
2 // Registered Incrementer
3 // ========================================================================
4 // This is a simple example of a module for a registered incrementer
5 // which combines a positive edge triggered register with a combinational
6 // +2 incrementer. We use flat register -transfer -level modeling.
7

8 `ifndef EX_REGINCR_REG_INCR_V
9 `define EX_REGINCR_REG_INCR_V

10

11 module ex_regincr_RegIncr
12 (
13 input logic clk ,
14 input logic [7:0] in,
15 output logic [7:0] out
16 );
17

18 // Sequential logic
19

20 logic [7:0] temp_reg;
21 always @( posedge clk )
22 temp_reg <= in;
23

24 // Combinational logic
25

26 logic [7:0] temp_wire;
27 always @(*)
28 temp_wire = temp_reg + 1;
29

30 // Combinational logic
31

32 assign out = temp_wire + 1;
33

34 endmodule
35

36 `endif /* EX_REGINCR_REG_INCR_V */

Code at https://github.com/cbatten/x/blob/master/ex-regincr-RegIncr.v

Figure 15: Registered Incrementer – An eight-bit registered +2 incrementer corresponding to the
diagram in Figure 14.

4.1. RTL Behavioral Model of Registered Incrementer

Figure 15 shows the Verilog code which corresponds to the diagram in Figure 14. Every Verilog file
should begin with a header comment as shown on Lines 1–6 in Figure 15. The header comment
should identify the primary module in the file, and include a brief description of what the module
does. Reserve discussion of the actual implementation for later in the file. In general, you should
attempt to keep lines in your Verilog source code to less than 74 characters. This will make your code
easier to read, enable printing on standard sized paper, and facilitate viewing two source files side-
by-side on a single monitor. Lines 8–9 create an “include guard” using the Verilog pre-processor. An
include guard ensures that even if we include this Verilog file multiple times the modules within the
file will only be declared once. Without include guards, the Verilog compiler will likely complain
that the same module has been declared multiple times. Make sure that you have the corresponding
end of the include guard at the bottom of your Verilog source file as shown on Line 36.

As always, we begin by identifying the module’s interface which in this case will include an eight-
bit input port, eight-bit output port, and a clock input. Lines 11–16 in Figure 15 illustrate how we
represent this interface using Verilog. A common mistake is to forget the semicolon (;) on Line 16. A
couple of comments about the coding conventions that we will be using in this course. All module
names should always include the subproject name as a prefix (e.g., ex_regincr_). The portion of
the name after this prefix should usually use CamelCaseNaming; each word begins with a capital
letter without any underscores (e.g., RegIncr). Port names (as well as variable and module instance
names) should use underscore_naming; all lowercase with underscores to separate words. As shown
on Lines 13–15, ports should be listed one per line with a two space initial indentation. The bitwidth
specifiers and port names should all line up vertically. As shown on Lines 12 and 16, the opening and
closing parenthesis should be on their own separate lines. Carefully group ports to help the reader
understand how these ports are related. Use port names (as well as variable and module instance
names) that are descriptive; prefer longer descriptive names (e.g., write_en) over shorter confusing
names (e.g., wen).

Lines 17–33 model the internal behavior of the module. We usually prefer using two spaces for each
level of indentation; larger indentation can quickly result in significantly wasted horizontal space.
You should always use spaces and never insert any real tab characters into your source code. You must
limit yourself to synthesizable RTL for modeling your design. We will exclusively use two kinds of
always blocks: always @( posedge clk ) concurrent blocks to model sequential logic and always
@(*) concurrent blocks to model combinational logic. We should always use the newer @(*) syn-
tax and never use explicit sensitivity lists. We require students to clearly distinguishing between
the portions of your code that are meant to model sequential logic from those portions meant to
model combinational logic. This simple guideline can save significant frustration by making it easy
to see subtle errors. For example, by convention we should only use non-blocking assignments in
sequential logic (e.g., the <= operator on Line 22) and we should only use blocking assignments in
combinational logic (e.g., the = operator on Line 28). We use the variable temp_reg to hold the inter-
mediate value between the register and the first +1 incrementer, and we use the variable temp_wire
to hold the intermediate value between the two +1 incrementers. temp_reg is modeling a register
while temp_wire is modeling a wire. Notice that both of these variables use the logic data type;
what makes one model a register while the other models a wire is how these variables are used.
The sequential concurrent block update to temp_reg means it models a register. The combinational
concurrent block update to temp_wire means it models a wire.

The register incrementer illustrates the two fundamental ways to model combinational logic. We
have used an always @(*) concurrent block to model the first +1 incrementer and a continuous as-
signment statement (i.e., assign) to model the second +1 incrementer. In general, we prefer continu-
ous assignment statements over always @(*) concurrent blocks to model combinational logic, since
it is easier to model less-realistic hardware using always @(*) concurrent blocks. There is usually
a more direct one-to-one mapping from continuous assignment statements to real hardware. How-
ever, there are many cases where it is significantly more convenient to use always @(*) concurrent
blocks or just not possible to use continuous assignment statements. Students will need to use their
judgment to determine the most elegant way to represent the hardware they are modeling while still
ensuring there is a clear mapping from the model to the target hardware.

Create a new Verilog source file named ex-regincr-RegIncr.v that contains the code in Figure 15
using your favorite text editor. You should place this file in the ex-regincr subproject and add it to
the list of files in the ex-regincr.mk makefile fragment like this:

% cd ${TUTROOT}/ex-regincr
% echo "
ex_regincr_deps =
ex_regincr_srcs = ex-regincr-RegIncr.v
ex_regincr_test_srcs =
ex_regincr_sim_srcs =
" > ex-regincr.mk

4.2. Verifying the Registered Incrementer Using Ad-Hoc Unit Testing

Now that we have developed a new hardware module, your first thought should always turn to
testing that module. Figure 16 shows an ad-hoc test using non-synthesizable Verilog. Note that we
must explicitly include any Verilog files which contain modules that we want to use; Line 7 includes
the Verilog source file that contains the registered incrementer. Lines 13–14 setup a clock with a
period of 10 time steps. Notice that we are assigning an initial value to the clk net on Line 13 and
then modifying this net every five timesteps; setting initial values such as this is not synthesizable
and should only be used in test harnesses. If you need to set an initial value in your design, you
should use properly constructed reset logic.

Lines 18–26 instantiate the device under test. Notice that we use underscore_naming for the module
instance name (e.g., reg_incr). You should almost always use named port binding (as opposed to
positional port binding) to connect nets to the ports in a module instance. Lines 23–25 illustrate the
correct coding convention with one port binding per line and the ports/nets vertically aligned. As
shown on Lines 22 and 26 the opening and closing parenthesis should be on their own separate lines.
Although this may seem verbose, this coding style can significantly reduce errors by making it much
easier to quickly visualize how ports are connected.

Lines 30–64 illustrate an initial block which executes at the very beginning of the simulation.
initial blocks are not synthesizable and should only be used in test harnesses. Lines 34–35 in-
struct the simulator to dump waveforms for all nets. Line 37 is a delay statement that essentially
waits for one timestep. Delay statements are not synthesizable and should only be used in test har-
nesses. Lines 41–46 implement a test by setting the inputs of the device under test, waiting for 10
time steps, and then checking that the output is as expected. If there is an error, we display an error
message and stop the simulation. We include two more tests, and if we make it to the bottom of the
initial block then the test has passed.

Create a new Verilog source file named ex-regincr-RegIncr-adhoc-test.v that contains the code
in Figure 16 using your favorite text editor. You should place this file in the ex-regincr subproject
and add it to the list of files in the ex-regincr.mk makefile fragment like this:

% cd ${TUTROOT}/ex-regincr
% echo "
ex_regincr_deps =
ex_regincr_srcs = ex-regincr-RegIncr.v
ex_regincr_test_srcs =
ex_regincr_sim_srcs = ex-regincr-RegIncr-adhoc-test.v
" > ex-regincr.mk

25



ECE 4750 Computer Architecture, Fall 2014 Tutorial 3: Verilog Hardware Description Language

1 // ========================================================================
2 // RegIncr Ad -Hoc Testing
3 // ========================================================================
4 // This is an example of ad -hoc testing which does not use any specific
5 // unit testing framework.
6

7 `include "ex-regincr -RegIncr.v"
8

9 module top;
10

11 // Clocking
12

13 logic clk = 1;
14 always #5 clk = ~clk;
15

16 // Instaniate the device -under -test
17

18 logic [7:0] in;
19 logic [7:0] out;
20

21 ex_regincr_RegIncr reg_incr
22 (
23 .clk (clk),
24 .in (in),
25 .out (out)
26 );
27

28 // Verify functionality
29

30 initial begin
31

32 // Dump waveforms
33

34 $dumpfile("ex -regincr -RegIncr -test.vcd");
35 $dumpvars;
36

37 #1;
38

39 // Test cases
40

41 in = 8'h00;
42 #10;
43 if ( out != 8'h02 ) begin
44 $display( "ERROR: out , expected = %x, actual = %x", 8'h02 , out );
45 $finish;
46 end
47

48 in = 8'h13;
49 #10;
50 if ( out != 8'h15 ) begin
51 $display( "ERROR: out , expected = %x, actual = %x", 8'h15 , out );
52 $finish;
53 end
54

55 in = 8'h27;
56 #10;
57 if ( out != 8'h29 ) begin
58 $display( "ERROR: out , expected = %x, actual = %x", 8'h29 , out );
59 $finish;
60 end
61

62 $display( "*** PASSED ***" );
63 $finish;
64 end
65

66 endmodule

Code at https://github.com/cbatten/x/blob/master/ex-regincr-RegIncr-adhoc-test.v

Figure 16: Ad-Hoc Test Harness for Registered Incrementer – An ad-hoc test harness for the eight-
bit registered incrementer in Figure 15.

We are now finally ready to test our registered incrementer module. You can build and run the
ad-hoc test as follows:

% cd ${TUTROOT}/build
% make ex-regincr-RegIncr-adhoc-test
% ./ex-regincr-RegIncr-adhoc-test

If you take a close look at the output from running make, you should see make using iverilog to
build the ad-hoc test. Essentially, make is automating the manual process we used in the previous
section. You might also note that make uses a helper script called warnings2errors. Because Verilog
is a very permissive language, iverilog will compile code which while technically allowed is almost
certainly wrong with just a warning. The warnings2errors helper script will convert all warnings
into errors and halt the compilation process. This forces the designer to eliminate all warnings and
should help avoid a large class of subtle bugs. If everything goes as expected, then the ad-hoc test
should display *** PASSED ***.

H To-Do On Your Own: Edit the register incrementer so that it now increments by +4 instead of +2.
Recompile, rerun the ad-hoc test, and verify that the tests no longer pass. Modify the ad-hoc test
so that it includes the correct reference outputs for a +4 incrementer, recompile, rerun the ad-hoc
test, and verify that the test now pass. When you are finished, edit the register incrementer so
that it again increments by +2.

4.3. Verifying the Registered Incrementer Using Unit Testing Framework

There are many issues with using ad-hoc test harnesses similar to what is shown in Figure 17. Ad-hoc
testing is verbose, which makes it error prone and more cumbersome to write tests. Ad-hoc testing
is difficult for others to read and understand since by definition it is ad-hoc. Ad-hoc testing does not
use any kind of standard test output, and does not provide support for controlling the amount of test
output. In this course, we will be using a simple unit testing framework to make it easier to write,
maintain, and run unit tests. This framework is in vc-test.v as part of the vc subproject.

Figure 17 shows how we can use this simple framework to unit test our registered incrementer.
Note that we need to explicitly include vc-test.v at the top of our unit test. All of our unit tests
should be in a single module named top. The VC_TEST_SUITE_BEGIN macro on Line 9 is used to set
the name of this test suite and initialize the framework; note that we must include a corresponding
VC_TEST_SUITE_END right before the end of the module.

Lines 11–54 illustrate a single test case; notice that all net names, module instance names, and tasks
are prefixed with t1 corresponding to test case number one. Our unit tests will usually include
several test cases, and each test case should use a unique prefix (e.g., t2, t3, etc.). The test case has
three parts: Lines 15–23 instantiate the device-under-test (DUT), Lines 27–39 declare a helper task,
and Lines 43–54 use the helper task to test the DUT. We should use a similar coding convention
as described above for instantiating the DUT. All input ports should be connected to reg nets and
all output ports should be connected to wire nets. The helper task should include one input for
each port of the DUT, and it should first set all of the inputs for the DUT and then use macros from
vc-test.v to verify that the outputs from the DUT are as expected. The delay statements on Lines 34,
37, and 46 are used to ensure that we set inputs and verify outputs right after the rising clock edge.
The VC_TEST_NOTE_INPUTS_1 macro is used to display the values of all inputs to the DUT and can
be useful when trying to debug failing test cases. There are corresponding macros which take two,
three, and four arguments. The VC_TEST_NET macro is used to verify that a net has an expected
value. The actual test case is declared using VC_TEST_CASE_BEGIN and VC_TEST_CASE_END macros.

1 // ========================================================================
2 // ex -regincr -RegIncr Unit Tests
3 // ========================================================================
4

5 `include "ex-regincr -RegIncr.v"
6 `include "vc-test.v"
7

8 module top;
9 `VC_TEST_SUITE_BEGIN( "ex-regincr -RegIncr" )

10

11 // ----------------------------------------------------------------------
12 // Test ex_regincr_RegIncr
13 // ----------------------------------------------------------------------
14

15 logic [7:0] t1_in;
16 logic [7:0] t1_out;
17

18 ex_regincr_RegIncr t1_reg_incr
19 (
20 .clk (clk),
21 .in (t1_in),
22 .out (t1_out)
23 );
24

25 // Helper task
26

27 task t1
28 (
29 input logic [7:0] in,
30 input logic [7:0] out
31 );
32 begin
33 t1_in = in;
34 #1;
35 `VC_TEST_NOTE_INPUTS_1( in );
36 `VC_TEST_NET( t1_out , out );
37 #9;
38 end
39 endtask
40

41 // Test case
42

43 `VC_TEST_CASE_BEGIN( 1, "simple" )
44 begin
45

46 #1;
47

48 t1( 8'h00 , 8'h?? );
49 t1( 8'h13 , 8'h02 );
50 t1( 8'h27 , 8'h15 );
51 t1( 8'hxx , 8'h29 );
52

53 end
54 `VC_TEST_CASE_END
55

56 `VC_TEST_SUITE_END
57 endmodule

Code at https://github.com/cbatten/x/blob/master/ex-regincr-RegIncr.t.v

Figure 17: Unit Tests for Registered Incrementer – A set of unit tests for the eight-bit registered
incrementer in Figure 15, which uses our vc-Test unit testing framework.

You must explicitly pass the test case number to VC_TEST_CASE_BEGIN, and a common mistake is
to use the same test case number more than once. We call the helper task once per line, and this
basically means each line of our test corresponds to setting the DUT inputs and verifying the DUT
outputs for one cycle. Note that you can use an x to indicate inputs which you do not care about (e.g.,
Line 51) and you can use a question mark (?) to indicate outputs for which any value is acceptable
(e.g., Line 48).

The unit testing in Figure 17 is an example of directed cycle-by-cycle gray-box testing. It is directed
since we are explicitly creating directed tests as opposed to using some kind of random testing. It is
cycle-by-cycle since we are explicitly setting the inputs and verifying the outputs every cycle. Black-
box testing describes a testing strategy where the test cases depend only on the interface and not the
specific implementation of the DUT (i.e., they should be valid for any correct implementation). White-
box testing describes a testing strategy where the test cases depend on the specific implementation of
the DUT (i.e., they may not be valid for every correct implementation). The test cases in Figure 17 are
black-box with respect to the functional behavior of the DUT but they are white-box with respect to the
timing behavior of the device. The test cases rely on the fact that the registered incrementer includes
exactly one edge and they would fail if we pipelined the incrementer such that each transaction took
two edges. In Section 6, we will see how we can use latency-insensitive interfaces to create true
black-box unit tests.

Create a new Verilog source file named ex-regincr-RegIncr.t.v that contains the code in Figure 17
using your favorite text editor. Note that all unit tests must use the .t.v filename extension. You
should place this file in the ex-regincr subproject and add it to the list of files in the ex-regincr.mk
makefile fragment like this:

% cd ${TUTROOT}/ex-regincr
% echo "
ex_regincr_deps =
ex_regincr_srcs = ex-regincr-RegIncr.v
ex_regincr_test_srcs = ex-regincr-RegIncr.t.v
ex_regincr_sim_srcs =
" > ex-regincr.mk

We can build and run our unit tests as follows:

% cd ${TUTROOT}/build
% make ex-regincr-RegIncr-test
% ./ex-regincr-RegIncr-test

Our simple unit testing framework provides support for verbose output which displays all tests even
if they pass with the +verbose command-line option. You can specify a specific test case to run with
the +test-case command-line option.

% cd ${TUTROOT}/build
% ./ex-regincr-RegIncr-test +verbose=1 +test-case=1

Developers will spend a great deal of time editing Verilog source code, rebuilding a unit test, and
then rerunning this unit test. We can simplify this edit-compile-test development cycle by building a
unit test and running that test on a single command line as follows:

% cd ${TUTROOT}/build
% make ex-regincr-RegIncr-test && ./ex-regincr-RegIncr-test

This way we can make an edit in our editor, change to our terminal, press the up-arrow key, and
press return. Remember that you can run all unit tests in a given subproject or all unit tests in the
entire project using the following make targets:

% cd ${TUTROOT}/build
% make ex-regincr-check
% make check

We recommend you spend most of your time building and running a single unit test when debugging
your hardware models, since this is faster and enables you to easily change the verbosity or run
a specific test case. Reserve the makefile check targets for when you want to quickly verify that
everything is working.

H To-Do On Your Own: Edit the register incrementer so that it now increments by +4 instead of +2.
Recompile, rerun the unit test, and verify that the tests no longer pass. Modify the unit test so
that it includes the correct reference outputs for a +4 incrementer, recompile, rerun the unit test,
and verify that the test now pass. When you are finished, edit the register incrementer so that it
again increments by +2.

4.4. Debugging the Registered Incrementer Using Waveforms

It is sometimes possible to debug your design purely using the output from your unit tests, but often
we need to inspect more signals than are displayed through the unit testing framework. While we
could add calls to the $display() system task to output more and more signals in our design, it is
much more productive to use the built-in capabilities of the Verilog tools to generate waveforms for
each net in your design. We will be using the Verilog Change Dump (VCD) format for our wave-
forms.

To have a unit test generate VCD, you need to pass it the +dump-vcd command line option. This will
cause the unit test framework to generate a VCD file with the same name as the test and a .vcd file
name extension.

% cd ${TUTROOT}/build
% ./ex-regincr-RegIncr-test +dump-vcd

You can use the open-source GTKWave program to browse the generated waveforms as follows:

% cd ${TUTROOT}/build
% gtkwave ex-regincr-RegIncr-test.vcd &

You can browse the module hierarchy of your design in the upper-left panel, with the signals in any
given module being displayed in the lower-left panel. Select signals and use the Append or Insert
button to add them to the waveform panel on the right. You can drag-and-drop signals to arrange
them as desired. To see the full hierarchical names of each signal choose Edit > Toggle Trace Hierarchy
or simply press the H key. Choose File > Reload Waveform (or press the blue circular arrows in the
toolbar) to update GTKWave after you have rerun a simulation. Organizing signals can sometimes
be quite time consuming, so you can save and load the current configuration using File > Write Save
File and File > Read Save File. Figure 18 illustrates using GTKWave to view the waveforms from
our ad-hoc test. GTKWave has many useful options which can make debugging your design more
productive, so feel free to explore the associated documentation.

Figure 18: GTKWave Waveform Viewer – GTKWave is being used to browse the signals associated
with the registered incrementer shown in Figure 15 and the ad-hoc test shown in Figure 17.

Note that viewing waveforms should never take the place of automated verification through the unit
testing framework. Students might be tempted to simply look at the waveforms to determine if their
design is working, but this kind of “verification by inspection” is error prone and not reproducible.
If you later make a change to your design, you would have to take another look at the waveforms
to ensure that your design still works. If another member of your group wants to understand your
design and verify that it is working, he or she would also need to take a look at the waveforms. While
this might be feasible for very simple designs, it is obviously not a scalable approach when building
the more complicated designs we will tackle in this course. Automated testing using the unit testing
framework is the best way to permanently verify your design.

H To-Do On Your Own: Edit the register incrementer so that it now increments by +4 instead of +2.
Recompile, rerun the unit test (with the +dump-vcd option), and take another look the waveforms
to see how they have changed.

5. Sorter: Structural Verilog Development and Verification

In this section, we will explore slightly more advanced techniques including parameterized design,
line tracing, structural composition, randomized testing, and evaluating the performance of a design
with a simulator. We will be using the simple pipelined four-element sorter shown in Figure 19 as
an example. The code related to this part of the tutorial is located in the ex-sorter subproject. Each
min/max unit compares its inputs and sends the smaller value to the top output port and the larger
value to the bottom output. This specific implementation is pipelined into three stages, such that the
critical path should be through a single min/max unit. Input and output valid signals indicate when
the input and output elements are valid.

Notice that we register the inputs but we do not register the outputs. In other words, we register
the inputs as soon as possible, but there is almost a full cycle’s worth of work before the outputs are
stable. When working with larger blocks we usually need to decide whether to use registered inputs

Figure 19: Block Diagram for Four-Element Sorter – An n-bit pipelined four-element sorter which
arranges the four elements in ascending order.

or registered outputs, and it is important that we adopt a uniform policy. When some blocks use
registered inputs and others use registered outputs, composing them can create either long critical
paths or “dead cycles” where very little work happens beyond simply transferring data. In this
course, we will adopt the general policy of using registered inputs for larger blocks. As long as all
modules roughly adhere to this policy then we can focus on the critical path of each larger module
in isolation and be confident that composing these blocks should not cause significant timing issues.

5.1. Flat Sorter Implementation

We will begin by exploring a flat implementation of the sorter that does not instantiate any additional
child modules. This implementation is provided for you in the file named ex-sorter-SorterFlat.v.
Figure 20 illustrates the Verilog code that describes the interface for the sorter. Notice how we have
parameterized the interface by the bitwidth of each element. Lines 2–4 declare a parameter named
p_nbits and give it a default value of one bit. We use this parameter when declaring the bitwidth of
the input and output ports, and we will also use this parameter in the implementation.

Figure 21 shows the first pipeline stage of the flat implementation of the sorter. Notice how we
use the parameter p_nbits to declare various internal registers and wires. We cleanly separate the
sequential logic from the combinational logic. We use comments and explicit suffixes to make it clear
what pipeline stage we are modeling.

The corresponding unit test is in ex-sorter-SorterFlat.t.v, although if you look in this file you
will see that it only contains a few lines. We are often evaluating several different implementations
of a module, and we would like to be able to easily run all of the same tests on all implementa-
tions. To reduce duplicate code, we have refactored the bulk of the test code into a single file called
ex-sorter-test-harness.v. We use a special pre-processor macro when instantiating the DUT
named EX_SORTER_IMPL in ex-sorter-test-harness.v. Then to test an implementation we sim-
ply define this special macro, include the Verilog file that contains the implementation, and include
the Verilog file that contains the test harness. You build and run the unit tests as follows:

% cd ${TUTROOT}/build
% make ex-sorter-SorterFlat-test && ./ex-sorter-SorterFlat-test

The design should pass all of the tests. We are using a very similar strategy as in the previous section
based on directed cycle-by-cycle gray-box testing. Since the sorter includes an output valid bit, it

1 module ex_sorter_SorterFlat
2 #(
3 parameter p_nbits = 1
4 )(
5 input logic clk ,
6 input logic reset ,
7

8 input logic in_val ,
9 input logic [p_nbits -1:0] in0 ,

10 input logic [p_nbits -1:0] in1 ,
11 input logic [p_nbits -1:0] in2 ,
12 input logic [p_nbits -1:0] in3 ,
13

14 output logic out_val ,
15 output logic [p_nbits -1:0] out0 ,
16 output logic [p_nbits -1:0] out1 ,
17 output logic [p_nbits -1:0] out2 ,
18 output logic [p_nbits -1:0] out3
19 );

Figure 20: Interface for the Four-Element
Sorter – The interface corresponds to the dia-
gram in Figure 19 and is parameterized by the
bitwidth of each element.

should be possible to use a black-box testing strategy by waiting for the valid bit to be true before
checking the outputs. This would allow the same test harness to test various implementations that
each take a different number of cycles to sort the elements (e.g., pipelined, iterative, single-cycle).

H To-Do On Your Own: The sorter currently sorts the four input numbers from smallest to largest.
Change to the sorter implementation so it sorts the numbers from largest to smallest. Recompile
and rerun the unit test and verify that the tests are no longer passing. Modify the tests so that
they correctly capture the new expected behavior. Make a copy of the sorter implementation file
so you can put things back to the way they were when you are finished.

5.2. Visualizing Execution with Line Traces

We have already learned about two methods to observe the operation of our device-under-test: the
test output from our unit testing framework and the waveforms for every net in the design. The
former is rather limited and the latter can sometimes be almost too detailed with bit accuracy and a
multitude of nets to understand. As architects, we often want a way to visualize the execution of our
device-under-test with an intermediate level of detail; for example, we want our models to generate
high-level pipeline diagrams to illustrate how transactions are moving through a system.

In this course, we will use a technique called line tracing to achieve this goal. A line trace consists
of plain-text trace output with each line corresponding to one (and only one!) cycle. Fixed-width
columns will correspond to either state at the beginning of the corresponding cycle or the output
of combinational logic during that cycle. Line traces will abstract the detailed bit representations of
signals in our design into useful character representations. So for example, instead of visualizing
instructions as raw bits, we will visualize them as text strings.

You can use the +trace command-line option to see a line trace of the four-element sorter.

% cd ${TUTROOT}/build
% ./ex-sorter-SorterFlat-test +test-case=1 +trace=1

Figure 22 shows a portion of the resulting line trace with some additional annotation. The first
column indicates the current cycle. There are fixed-width columns showing the inputs and outputs
of the module along with the state in the S0, S1, and S2 pipeline registers at the beginning of the cycle.
Notice that we use spaces when the data is invalid which improves readability and makes it easy to

1 // ----------------------------------------------------------------------
2 // Stage S0->S1 pipeline registers
3 // ----------------------------------------------------------------------
4

5 logic val_S1;
6 logic [p_nbits -1:0] elm0_S1;
7 logic [p_nbits -1:0] elm1_S1;
8 logic [p_nbits -1:0] elm2_S1;
9 logic [p_nbits -1:0] elm3_S1;

10

11 always @( posedge clk ) begin
12 val_S1 <= (reset) ? 0 : in_val;
13 elm0_S1 <= in0;
14 elm1_S1 <= in1;
15 elm2_S1 <= in2;
16 elm3_S1 <= in3;
17 end
18

19 // ----------------------------------------------------------------------
20 // Stage S1 combinational logic
21 // ----------------------------------------------------------------------
22 // Note that we explicitly catch the case where the elements contain
23 // X's and propagate X's appropriately. We would not need to do this if
24 // we used a continuous assignment statement with a ternary conditional
25 // operator.
26

27 logic [p_nbits -1:0] elm0_next_S1;
28 logic [p_nbits -1:0] elm1_next_S1;
29 logic [p_nbits -1:0] elm2_next_S1;
30 logic [p_nbits -1:0] elm3_next_S1;
31

32 always @(*) begin
33

34 // Sort elms 0 and 1
35

36 if ( elm0_S1 <= elm1_S1 ) begin
37 elm0_next_S1 = elm0_S1;
38 elm1_next_S1 = elm1_S1;
39 end
40 else if ( elm0_S1 > elm1_S1 ) begin
41 elm0_next_S1 = elm1_S1;
42 elm1_next_S1 = elm0_S1;
43 end
44 else begin
45 elm0_next_S1 = 'hx;
46 elm1_next_S1 = 'hx;
47 end
48

49 // Sort elms 2 and 3
50

51 if ( elm2_S1 <= elm3_S1 ) begin
52 elm2_next_S1 = elm2_S1;
53 elm3_next_S1 = elm3_S1;
54 end
55 else if ( elm2_S1 > elm3_S1 ) begin
56 elm2_next_S1 = elm3_S1;
57 elm3_next_S1 = elm2_S1;
58 end
59 else begin
60 elm2_next_S1 = 'hx;
61 elm3_next_S1 = 'hx;
62 end
63

64 end

Figure 21: First Stage of the Flat Sorter Implementation – First pipeline stage of the sorter using a
flat implementation corresponding to the diagram in Figure 19.

1 cycle inputs stage S0 stage S1 stage S2 outputs
2

3 12: {00,01,00,01}| | | |
4 13: |{00,01,00,01}| | |
5 14: | |{00,01,00,01}| |
6 15: | | |{00,01,00,01}|{00,00,01,01}
7 16: {04,03,02,01}| | | |
8 17: |{04,03,02,01}| | |
9 18: {05,07,06,08}| |{03,04,01,02}| |

10 19: |{05,07,06,08}| |{01,02,03,04}|{01,02,03,04}
11 20: | |{05,07,06,08}| |
12 21: | | |{05,07,06,08}|{05,06,07,08}
13 22: {04,03,02,01}| | | |
14 23: {a5,a3,a2,a7}|{04,03,02,01}| | |
15 24: {05,07,06,08}|{a5,a3,a2,a7}|{03,04,01,02}| |
16 25: {b0,b1,b0,b1}|{05,07,06,08}|{a3,a5,a2,a7}|{01,02,03,04}|{01,02,03,04}
17 26: {c7,c1,c2,c6}|{b0,b1,b0,b1}|{05,07,06,08}|{a2,a5,a3,a7}|{a2,a3,a5,a7}

Figure 22: Example Line Trace for Sorter – Each line corresponds to one (and only one!) cycle, and
the fixed-width columns correspond to either the state at the beginning of the corresponding cycle
or the output of combinational logic during that cycle.

1 `ifndef SYNTHESIS
2

3 reg [(( `VC_TRACE_NBITS_NCHARS(p_nbits )*4+5)*8) -1:0] str;
4

5 `VC_TRACE_BEGIN
6 begin
7

8 // Inputs
9

10 $sformat( str , "{%x,%x,%x,%x}", in0 , in1 , in2 , in3 );
11 vc_trace.append_val_str( trace_str , in_val , str );
12 vc_trace.append_str( trace_str , "|" );
13

14 ...
15

16 end
17 `VC_TRACE_END
18

19 `endif /* SYNTHESIS */

Figure 23: Example of Line Tracing Code – This code generates the first fixed-width column for the
line trace shown in Figure 22

see when the hardware is actually doing useful work. The line trace shows the sorter working on a
single sort and then trying multiple sorts at once in a pipelined fashion. You can see how each stage
is swapping various elements, and you can see that if we push in elements fast enough we are able
to keep the pipeline busy without any stalls. Line traces are a powerful way to visualize your design
and debug both correctness and performance issues.

The VC library includes support in vc-trace.v to make it easier to create line traces in your own
Verilog modules. Figure 23 shows a small snippet of code that is used in the sorter implementation
to trace the input ports. Notice how we have wrapped the line tracing code in ‘ifndef SYNTHESIS
and ‘endif to clearly indicate that this code is not synthesizable even though it is included in out
design. Line 3 declares a temporary string variable that we will use when converting nets into
strings. Lines 5–17 use helper macros to declare a task called trace which takes a special argu-
ment called trace_str that holds the current line trace string. The job of this task is to append

trace information to the trace_str that describes the current state and operation of this module.
The vc_trace.append_str and vc_trace.append_val_str helper tasks add strings to the line trace.
You can also build line traces hierarchically by explicitly calling the trace task on a child module.
Although we will provide significant line tracing code in each lab harness, you are also strongly
encouraged to augment this code with your own line tracing.

H To-Do On Your Own: Modify the line tracing code to show the pipeline stage label (in, S0, S1, S2,
out) before each stage. So cycle 24 of the line trace shown in Figure 22 should now look like:

24: in:{05,07,06,08}|S0:{a5,a3,a2,a7}|S1:{03,04,01,02}|S2: |out:

5.3. Structural Sorter Implementation

The flat implementation in ex-sorter-SorterFlat.v is complex and monolithic and it fails to really
exploit the structure inherent in the sorter. We can use modularity and hierarchy to divide compli-
cated designs into smaller more manageable units; these smaller units are easier to design and can be
tested independently before integrating them into larger, more complicated designs. We have started
a structural implementation in ex-sorter-SorterStruct.v; the structural implementation will have
an identical interface and behavior as the flat implementation in ex-sorter-SorterFlat.v.

Figure 24 shows the first pipeline stage of the structural implementation of the sorter. Our design
instantiates three kinds of modules: vc_ResetReg, vc_Reg, and ex_sorter_MinMaxUnit. The register
modules are provided in the VC library. Notice how we still use the parameter p_nbits to declare
various internal variables, but in addition, we use this parameter when instantiating parameterized
sub-modules. For example, the vc_Reg module is parameterized, and this allows us to easily create
pipeline registers of any bitwidth. Even though we are using a structural implementation strategy,
we still cleanly separate the sequential logic from the combinational logic. We still use comments
and explicit suffixes to make it clear what pipeline stage we are modeling.

H To-Do On Your Own: The structural implementation is incomplete because the actual implementa-
tion of the MinMax unit in vc-MinMaxUnit.v is not finished. You should go ahead and implement
the MinMax unit, and then as always you should write a unit test to verify the functionality of your Min-
Max unit! You should have enough experience based on the previous sections to be able to create
a unit test from scratch and add it to the build system. Once your MinMax unit is complete and
tested, then test the structural sorter implementation like this:

% cd ${TUTROOT}/build
% make ./ex-sorter-SorterStruct-test && ./ex-sorter-SorterStruct-test

1 // ----------------------------------------------------------------------
2 // Stage S0->S1 pipeline registers
3 // ----------------------------------------------------------------------
4

5 logic val_S1;
6

7 vc_ResetReg #(1) val_S0S1
8 (
9 .clk (clk),

10 .reset (reset),
11 .d (in_val),
12 .q (val_S1)
13 );
14

15 // This is probably the only place where it might be acceptable to use
16 // positional port binding since (a) it is so common and (b) there are
17 // very few ports to bind.
18

19 logic [p_nbits -1:0] elm0_S1;
20 logic [p_nbits -1:0] elm1_S1;
21 logic [p_nbits -1:0] elm2_S1;
22 logic [p_nbits -1:0] elm3_S1;
23

24 vc_Reg #( p_nbits) elm0_S0S1( clk , elm0_S1 , in0 );
25 vc_Reg #( p_nbits) elm1_S0S1( clk , elm1_S1 , in1 );
26 vc_Reg #( p_nbits) elm2_S0S1( clk , elm2_S1 , in2 );
27 vc_Reg #( p_nbits) elm3_S0S1( clk , elm3_S1 , in3 );
28

29 // ----------------------------------------------------------------------
30 // Stage S1 combinational logic
31 // ----------------------------------------------------------------------
32

33 logic [p_nbits -1:0] mmuA_out_min_S1;
34 logic [p_nbits -1:0] mmuA_out_max_S1;
35

36 ex_sorter_MinMaxUnit #( p_nbits) mmuA_S1
37 (
38 .in0 (elm0_S1),
39 .in1 (elm1_S1),
40 .out_min (mmuA_out_min_S1),
41 .out_max (mmuA_out_max_S1)
42 );
43

44 logic [p_nbits -1:0] mmuB_out_min_S1;
45 logic [p_nbits -1:0] mmuB_out_max_S1;
46

47 ex_sorter_MinMaxUnit #( p_nbits) mmuB_S1
48 (
49 .in0 (elm2_S1),
50 .in1 (elm3_S1),
51 .out_min (mmuB_out_min_S1),
52 .out_max (mmuB_out_max_S1)
53 );

Figure 24: First Stage of the Structural Sorter Implementation – First pipeline stage of the sorter
using a structural implementation corresponding to the diagram in Figure 19.

5.4. Verifying the Sorter Using Randomized Testing

So far we have mostly used a directed cycle-by-cycle gray-box testing strategy. Once we have fin-
ished writing hand-crafted directed tests, we almost always want to leverage randomized testing to
further improve our confidence in the correct functionality of the design. Generating randomized test
vectors in Verilog can be cumbersome, and indeed there are sophisticated programming languages
whose sole purpose it is to facilitate more productive testing of hardware models. SystemVerilog
adds comprehensive support for randomized testing, but these features are not yet supported in
iverilog. In this course, we will take a simpler approach based on using Python to generate ran-
domized test inputs and the corresponding correct outputs for verification. The build system simpli-
fies creating Python scripts that output Verilog code to be included in our unit tests.

We have already included such a Python script in the ex-sorter subproject. Try running the script
like this:

% cd ${TUTROOT}/build
% python ../ex-sorter/ex-sorter-gen-input.py random

Spend a few minutes looking at the source code for the Python script. The result will be a hundred
test vectors written in a very similar style to what have been using in our directed tests. After writing
a Python script like this, we need to tell the build system about the Verilog file we want to generate us-
ing this script. You can do this by adding a specially named Verilog file to a new make variable in the
subprojects make fragment. Take a look at how the ex-sorter.mk file uses the ex_sorter_pyv_src
make variable to specify three different Python-generated Verilog files. The build system uses the
filename to figure out that to generate the Verilog file named ex-sorter-gen-input_random.py.v
we need to run the ex-sorter-gen-input.py Python script with random as a command line argu-
ment. After updating the ex-sorter.mk makefile fragment you should be able to generate these
Verilog files like this:

% cd ${TUTROOT}/build
% make ex-sorter-gen-input_random.py.v

We can include these generated Verilog files in dedicated test cases. Figure 25 illustrates a new test
case we can add to ex-sorter-test-harness.v for a randomized testing strategy. Adding this new

1 // ----------------------------------------------------------------------
2 // Test Random
3 // ----------------------------------------------------------------------
4

5 `VC_TEST_CASE_BEGIN( 2, "ex-sorter -SorterFlat (random data)" )
6 begin
7

8 #1; t1_reset = 1'b1;
9 #20; t1_reset = 1'b0;

10

11 `include "ex-sorter -gen -input_random.py.v"
12

13 end
14 `VC_TEST_CASE_END

Code at https://github.com/cbatten/x/blob/master/code-tut3-random-testing.v

Figure 25: Randomized Test Case for the Four-Element Sorter – The Verilog file
ex-sorter-gen-input_random.py.v is generated by a Python script and contains 100 random
test inputs and the corresponding correct outputs.

test case to the directed tests in ex-sorter-test-harness.v for both the flat and structural sorter
implementations.

When you are done you can run all of the tests for our ex-sorter subproject like this:

% cd ${TUTROOT}/build
% make ex-sorter-check

H To-Do On Your Own: Experiment with the two other random test vectors that the Python script
can generate. Add two more random test cases to ex-sorter-test-harness.v which apply these
two additional random test vectors to the sorter implementations. Recompile and rerun the unit
tests.

5.5. Evaluating the Sorter Using a Simulator

So far we have focused on implementing and verifying our design, but our ultimate goal is to actu-
ally evaluate a design. We do not use unit tests for evaluation; instead we use a evaluation simulator
which has been designed for quantitatively measuring the cycle-level performance of a specific im-
plementation on a given input dataset. For this tutorial, we will create a simulator to compare the flat
and structural implementations on the random datasets generated using the Python script described
in the previous section.

Due to limitations of the Verilog language, we will need a separate evaluation simulator binary for
every implementation we wish to evaluate. In order to reduce duplicate code, we refactor most of
the code into a shared simulator harness which is analogous to the technique we used in creating
a shared test harness. This is illustrated in the ex-sorter-sim-harness.v, ex-sorter-sim-flat.v,
ex-sorter-sim-struct.v files. The simulator harness uses a special pre-processor macro when in-
stantiating the sorter (i.e., EX_SORTER_IMPL in ex-sorter-sim-harness.v). The simulator harness is
responsible for handling command line arguments, instantiating the design, clocking the design, and
of course setting up the specific input dataset we wish to use in our evaluation. A key difference be-
tween a simulator and a unit test, is that the simulator should also report various statistics that help
us evaluate our design. The actual simulator files are relatively simple (i.e., ex-sorter-sim-flat.v,
ex-sorter-sim-struct.v); they define the special implementation macro, include the implementa-
tion, and include the harness. We have already added the harness and simulator Verilog source files
to the appropriate make variables in the subproject make fragment.

We can build and run a simulator using the following commands:

% cd ${TUTROOT}/build
% make ex-sorter-sim-flat
% ./ex-sorter-sim-flat +input=random +trace=1 +stats

The simulator should print out how many cycles it takes on average to sort four elements. Not
surprisingly, it should take one cycle since we are using a fully pipelined implementation. The sub-
project’s makefile fragment includes special rules to run a selection of experiments using different
implementations and input data sets. You can use the following to run all of these simulations and
report the results:

% cd ${TUTROOT}/build
% make ex-sorter-eval

The output should confirm that the flat and structural implementations have the same performance
on all input datasets. Note that if you have not actually completed the real implementation of the
MinMax unit, the simulations will still run and actually they will also report what looks to be reason-
able performance results; even though the structural implementation is not at all functionally correct. The
take-away here is that you should not use the simulator and/or evaluation for verification; your test-
ing strategy should be comprehensive enough that once you get to the evaluation you are confident
that your design is fully functional.

6. Greatest Common Divisor: Verilog Design Example

In the final section of the tutorial, we will leverage what we learned in the previous sections to study
a simple hardware unit that calculates the greatest common divisor (GCD) of two input operands.
The code related to this part of the tutorial is located in the ex-gcd subproject. Figure 26 illustrates
the interface for our module. The GCD unit will take two n-bits operands and produce an n-bit
result. For flow-control we use a latency-insensitive val/rdy interface on both the input and output.
Our implementation will use Euclid’s algorithm to iteratively calculate the GCD. Figure 27 illustrates
this algorithm as an executable Python function.

H To-Do On Your Own: Experiment with the algorithm using a Python interpreter. Try calculating
the GCD for different input values. Add additional debugging output to track what the algorithm
is doing each iteration.

6.1. Latency-Insensitive Message-Based Interfaces

The interface for the registered incrementer in Section 4 included no extra control signals. A module
which wants to use the registered incrementer must explicitly handle the fact that the unit always
takes exactly one cycle. The interface for the sorter in Section 5 included an extra valid signal. A
module which wants to use the sorter could be carefully constructed so as to be agnostic to the la-
tency of the sorter; this would enable flexibly trying out different sorting algorithms. One issue with
including just a valid signal is that there is no way to know if the sorter is busy, and there is no way
to tell the sorter that we are not ready to accept the result. In other words, there is no provision for
back pressure. As shown in Figure 26, our GCD design will use a fully latency-insensitive interface by
including two extra signals: a valid and a ready signal. These signals will allow additional flexibil-

req_msg

req_val

req_rdy

resp_msg

resp_val

resp_rdy

Figure 26: Functional-Level Implementation of
GCD Unit – Input and output use latency-
insensitive val/rdy interfaces. The input message
includes two 16-bit operands; output message is an
16-bit result. Clock and reset signals are not shown.

1 def gcd( a, b ):
2 while True:
3 if a < b:
4 a,b = b,a
5 elif b != 0:
6 a = a - b
7 else:
8 return a

Figure 27: Euclid’s GCD Algorithm – Iter-
atively subtract the smaller value from the
larger value until one of them is zero, at
which time the GCD is the non-zero value.
This is executable Python code.

ity: the GCD unit can indicate it is not ready to accept a new GCD input, and another module can
indicate that it is not ready to accept the GCD output.

Assume we have a producer that wishes to send a message to a consumer using the val/rdy micro-
protocol. At the beginning of the cycle, the producer determines if it has a new message to send to
the consumer. If so, it sets the message bits appropriately and then sets the valid signal high. Also
at the beginning of the cycle, the consumer determines if it is able to accept a new message from the
producer. If so, it sets the ready signal high. At the end of the cycle, the producer and consumer can
independently AND the valid and ready signals together; if both signals are true then the message
is considered to have been sent from the producer to the consumer and both sides can update their
internal state appropriately. Otherwise, we will try again on the next cycle. To avoid long combina-
tional paths and/or combinational loops, we should avoid making the valid signal depend on the
ready signal or the ready signal depend on the valid signal. If you absolutely must, you can make
the ready signal depend on the valid signal (e.g., in an arbiter) but it is considered very bad practice
to make the valid signal depend on the ready signal. As long as you adhere to this policy, composing
modules via the val/rdy interface should not cause significant timing issues.

Based on the discussion so far, the benefit of a latency-insensitive val/rdy interface should be ob-
vious. This interface will allow true black-box testing and will allow flexibly composing modules
without regards for the detailed timing properties of each module. For example, if we use the GCD
unit in a larger design we can later decide to try a different GCD implementation (with potentially a
very different latency), and the larger design should need no modifications! We will use this kind of
interface extensively throughout the course.

To complement the val/rdy microprotocol, we can use struct types to encapsulate the actual mes-
sages we send over a val/rdy interface. Take a look at the two struct types and corresponding unit
test in ex-gcd-msgs.v and ex-gcd-msgs.t.v. We will use similar struct types for memory messages
and network packets. We can use a unit test to verify that the message type is written correctly:

% cd ${TUTROOT}/build
% make ex-gcd-msgs-test && ./ex-gcd-msgs-test

H To-Do On Your Own: Add a few more tests to the ex_gcd_req_msg_t test case in ex-gcd-msgs.t.v
and verify that these values are correctly written into and read from the new struct type.

6.2. Functional-Level Modeling

Figure 26 illustrates how we might provide a functional-level model for our GCD unit. RTL models are
synthesizable and meant to accurately model the target hardware, while a functional-level model is
simply meant to accurately capture the input/output functional behavioral of the target hardware
without any timing. A functional-level implementation of GCD is provided in ex-gcd-GcdUnitFL.v
and a portion of this implementation is shown in Figure 28.

It should be clear from the code that this implementation is in no way synthesizable. Functional
models often push most of the logic into a single large always @( posedge clk ) concurrent block
and make use of blocking assignments within this block to enable “procedural programming” similar
to what you might use in a traditional imperative programming language (e.g., C or Python). The
functional model actually uses Euclid’s algorithm to calculate the GCD, and this requires a data-
dependent while loop which is also obviously not synthesizable. Since this is just a functional model,
we are free to use any algorithm we wish as long as it has the correct input/output functional-level
behavior.

1 // ----------------------------------------------------------------------
2 // Implement GCD with Euclid 's Algorithm
3 // ----------------------------------------------------------------------
4

5 logic [`EX_GCD_REQ_MSG_A_NBITS -1:0] A;
6 logic [`EX_GCD_REQ_MSG_B_NBITS -1:0] B;
7 logic [`EX_GCD_RESP_MSG_RESULT_NBITS -1:0] temp;
8

9 logic full , req_go , resp_go , done;
10

11 always @( posedge clk ) begin
12

13 // Ensure that we clear the full bit if we are in reset.
14

15 if ( reset )
16 full = 0;
17

18 // At the end of the cycle , we AND together the val/rdy bits to
19 // determine if the input/output message transactions occured.
20

21 req_go = req_val && req_rdy;
22 resp_go = resp_val && resp_rdy;
23

24 // If the output transaction occured , then clear the buffer full bit.
25 // Note that we do this _first_ before we process the input
26 // transaction so we can essentially pipeline this control logic.
27

28 if ( resp_go )
29 full = 0;
30

31 // If the input transaction occured , then write the input message
32 // into our internal buffer and update the buffer full bit.
33

34 if ( req_go ) begin
35 A = req_msg.a;
36 B = req_msg.b;
37 full = 1;
38 end
39

40 // The output message is always the GCD of the buffer
41

42 done = 0;
43 while ( !done ) begin
44 if ( A < B ) begin
45 temp = A;
46 A = B;
47 B = temp;
48 end
49 else if ( B != 0 )
50 A = A - B;
51 else
52 done = 1;
53 end
54

55 resp_msg.result <= A;
56

57 // The output message is valid if the buffer is full
58

59 resp_val <= full;
60

61 end
62

63 // Connect output ready signal to input to ensure pipeline behavior
64

65 assign req_rdy = resp_rdy;

Figure 28: Part of the Functional-Level Implementation GCD Unit – This code corresponds to the
diagram in Figure 26 and implements the algorithm in Figure 27 using functional-level Verilog. This
model is definitely not synthesizable.

Functional-level models will be critical for enabling the verification lead to start verification tasks
early in the project. The verification lead can write and check tests using the functional-level model,
and then gradually these same tests can be used with the RTL implementation. Using the functional-
level model to write tests also ensures if the RTL implementation fails a test it is more likely due to
the RTL implementation itself as opposed to an incorrect test case.

6.3. Verifying GCD Using Test Sources and Sinks

One of the nice features of using a latency-insensitive val/rdy interface is that it enables us to use a
common framework for sending messages into the DUT and then verifying that the correct messages
come out of the DUT. Our VC library includes vc_TestRandDelaySource and vc_TestRandDelaySink
for this purpose, and we leverage these modules in building the unit tests for our functional-level
GCD implementation. Take a closer look at the test harness in ex-gcd-test-harness.v. We instan-
tiate a test source and attach it to the GCD unit’s input val/rdy interface, and then we instantiate a
test sink and attach it to the GCD unit’s output val/rdy interface. Figure 29 illustrates the overall
connectivity in the test harness. The test source includes the ability to randomly delay messages
going into the DUT and the test sink includes the ability to randomly apply back-pressure to the
DUT. By using various combinations of these random delays we can more robustly ensure that our
flow-control logic is working correctly. Note that these unit tests illustrate both directed black-box and
randomized black-box testing strategies. A similar approach of using Python-generated Verilog is used
to create the random source/sink messages for randomized testing.

A common testing strategy is for the very first test-case to use directed source/sink messages with no
random delays. For example, the first test case of our functional-level GCD implementation creates
a couple of source messages and then manually we have determined the correct sink messages.

% cd ${TUTROOT}/build
% make ex-gcd-GcdUnitFL-test
% ./ex-gcd-GcdUnitFL-test +test-case=1 +trace=1

Recall that you can generate a VCD file to view waveforms using the +dump-vcd command-line
option.

% cd ${TUTROOT}/build
% ./ex-gcd-GcdUnitFL-test +test-case=1 +dump-vcd
% gtkwave ex_gcd_GcdUnitFL-test.vcd

Once we know that our design works without any random delays, we continue to use directed
source/sink messages but then add random source delays and random sink delays. For example,
the second test case of our functional-level GCD implementation sets the test source to randomly

GCD Unit

req_msg

req_val

req_rdy

resp_msg

resp_val

resp_rdy
Test Source Test Sink

Figure 29: Verifying GCD Using Test Sources and Sinks – Parameterized test sources send a stream
of messages over a val/rdy interface, and parameterized test sinks receive a stream of messages over
a val/rdy interface and compare each message to a previously specified reference message. Clock
and reset signals are not shown.

1 cycle src A B out sink
2

3 19: 0004001c > 001c:0004(). > .
4 20: > ()# > #
5 21: > ()# > #
6 22: # > # ()# > #
7 23: # > # ()# > #
8 24: # > # ()# > #
9 25: 00990096 > 0096:0099()0004 > 0004

10 26: > ()# > #
11 27: > ()# > #
12 28: 009c00b4 > 00b4:009c()0003 > 0003
13 29: . > . ()000c > 000c
14 30: 00a00060 > 0060:00a0(). > .
15 31: # > # ()# > #
16 32: 005400a4 > 00a4:0054()0020 > 0020
17 33: # > # ()# > #
18 34: 00ab0059 > 0059:00ab()0004 > 0004

Figure 30: Line Trace for Functional-Level
Implementation of GCD – . +indicates the
val/rdy interface is not valid and not ready; # =
val/rdy interface is valid but not ready; space =
val/rdy interface is not valid and ready; mes-
sage is shown when it is actually transferred
across interface.

delay the input messages from zero to five cycles. We can also try using no delays on the source, but
adding random delays to the sink, and finally add random delays to both the source and the sink. If
we see that our design passes the tests with no random delays but fails with random delays this is a
good indicator that there is an issue with our val/rdy logic.

% cd ${TUTROOT}/build
% ./ex-gcd-GcdUnitFL-test +test-case=2 +trace=1
% ./ex-gcd-GcdUnitFL-test +test-case=3 +trace=1
% ./ex-gcd-GcdUnitFL-test +test-case=4 +trace=1

After additional directed testing with random delays, we can start to use randomly generated source/sink
messages for even greater test coverage.

% cd ${TUTROOT}/build
% ./ex-gcd-GcdUnitFL-test +test-case=5 +trace=1

Figure 30 illustrates a portion of the line trace for the randomized testing. Notice that the line trace
tells something about what is going on with each val/rdy interface. A period (.) indicates that the
interface is not ready but also not valid; a hash (#) indicates that the interface is valid but not ready;
a space indicates that the interface is ready but not valid. The actual message is displayed when it
is transferred from the producer to the consumer. We can see a message being sent into the GCD
unit on cycle 19 and although the result is valid on cycle 20 the test sink is not ready until cycle 25 to
accept the result. On cycles 20–21 the test source does not have a new message to send to the GCD
unit. On cycle 22 it does indeed have a new message, but the GCD unit is not ready because it is still
waiting on the test sink. Finally, on cycle 25 the test sink is ready and the GCD unit is able to send
the result and accept a new input. The GCD unit takes a single cycle when there is no back pressure;
we can see this on cycle 28–29.

H To-Do On Your Own: Write a new task to generate source/sink messages inspired by the
init_simple tasks currently in ex-gcd-test-harness.v. Your new task should be called
init_coprime and it should include a few sets of relatively prime numbers. Two numbers are
relatively prime (or coprime) if their greatest common divisor is one. Once you have defined this
new task, create a new test case using the VC_TEST_CASE_BEGIN and VC_TEST_CASE_END macros.
Your first new test case should set the random delays to zero, call your new init_coprime task,
and then call the run_test task. Once this new test case is working, add more test cases that use
the init_coprime task and change the random delays.

6.4. Control/Datapath Split Implementation

When modeling more complicated hardware, we will usually divide the design into two parts: the
datapath and the control unit. The datapath contains the arithmetic operators, muxes, and regis-
ters that work on the data, while the control unit is responsible for controlling these components
to achieve the desired functionality. The control unit sends control signals to the datapath and the
datapath sends status signals back to the control unit. Figure 31 illustrates the datapath for the GCD
unit and Figure 32 illustrates the corresponding finite-state-machine (FSM) control unit. The Ver-
ilog code for the datapath, control unit, and the top-level module which composes the datapath and
control unit is in ex-gcd-GcdUnit.v.

Take a look at the datapath interface which is also shown in Figure 33. Notice how we use the
GCD message types (i.e., ex_gcd_req_msg_t, ex_gcd_resp_msg_t) when specifying the input and
output ports. We have encapsulated all of the control signals in their own struct type, and also
all of the status signals in their own struct type. This approach enables compactly declaring the
control/status signal ports in the datapath and control unit, and also enables compactly connecting
the datapath and control unit together. Even more important, if we need to make a change or add a
new control/status signal we only need to modify one place in our code instead of three.

Figure 31: Datapath Diagram for GCD – Datapath in-
cludes two state registers and required muxing and arith-
metic units to iteratively implement Euclid’s algorithm.

Figure 32: FSM Diagram for GCD – A
hybrid Moore/Mealy FSM for control-
ling the datapath in Figure 31. Mealy
transitions in the calc state determine
whether to swap or subtract.

Figure 33 also shows the first two datapath components, but take a look in ex-gcd-GcdUnit.v to
see the entire datapath. Notice how we use a very structural implementation which exactly matches
the datapath diagram in Figure 31. We leverage several modules from the VC library (e.g., vc_Mux2,
vc_ZeroComparator, vc_Subtractor). You should use a similar structural approach when building
your own datapaths for this course. For a net that moves data from left to right in the datapath
diagram, we usually declare a dedicated wire right before the module instance (e.g., a_mux_out and
a_reg_out). For a net that moves data from right to left in the datapath diagram, we need to declare
a dedicated wire right before it is used as an input (e.g., b_reg_out and b_sub_out).

Take a look at the control unit and notice the stylized way we write FSMs. An FSM-based control
unit should have three parts: a sequential concurrent block for the state, a combinational concurrent
block for the state transitions, and a combinational concurrent block for the state outputs. We often
use case statements to compactly represent the state transitions and outputs. Figure 34 shows the
portion of the FSM responsible for setting the output signals. We use a task to set all of the control
signals in a single line; as long as the task does not include non-synthesizable constructs (e.g., delay
statements or system tasks) the task itself should be synthesizable. Essentially, we have created a
“control signal table” in our Verilog code which exactly matches what we might draw on a piece of
paper. There is one row for each state or Mealy transition and one column for each control signal.
These compact control signal tables simplify coding complicated FSMs (or indeed other kinds of
control logic) and can enable a designer to quickly catch bugs (e.g., are the enable signals always set
to either zero or one?).

The RTL implementation of the GCD unit uses the exact same test setup as for the functional-level
implementation, even though the functional-level implementation always takes a single cycle while
the RTL implementation takes a variable number of cycles; this illustrates the power of using latency-
insensitive interfaces. Let’s run the unit tests for the GCD implementation.

% cd ${TUTROOT}/build
% make ex-gcd-GcdUnit-test && ./ex-gcd-GcdUnit-test +trace=1

Figure 35 illustrates a portion of the line trace for the randomized testing. We use the line trace to
show the state of the A and B registers at the beginning of each cycle and use specific characters to
indicate which state we are in (i.e., I = idle, Cs = calc with swap, C- = calc with subtract, D = done).
We can see that the test source sends a new message into the GCD unit on cycle 296. The GCD unit
is in the idle state and transitions into the calc state. It does five subtractions and a final swap before
transitioning into the done state on cycle 304. The result is valid but the test sink is not ready, so the
GCD unit waits in the done state until cycle 310 when it is able to send the result to the test sink. On
cycle 311 the GCD unit accepts a new input message to work on. This is a great example of how an
effective line trace can enable you to quickly visualize how a design is actually working.

H To-Do On Your Own: Write a new task to generate source/sink messages inspired by the
init_simple tasks currently in ex-gcd-test-harness.v. Your new task should be called
init_same and it should include a few sets of numbers where both operands are exactly the same.
Once you have defined this new task, create a new test case using the VC_TEST_CASE_BEGIN and
VC_TEST_CASE_END macros. Your new test case should set the random delays to zero, call your
new init_coprime task, and then call the run_test task. Use the functional-level model to ver-
ify that your test case is correct before testing the RTL implementation. Use the line trace to
understand how these kinds of inputs perform.

1 // Control signals (ctrl ->dpath)
2

3 typedef struct packed {
4 logic a_reg_en; // Enable for A register
5 logic b_reg_en; // Enable for B register
6 logic [1:0] a_mux_sel; // Sel for mux in front of A reg
7 logic b_mux_sel; // sel for mux in front of B reg
8 } ex_gcd_cs_t;
9

10 // Status signals (dpath ->ctrl)
11

12 typedef struct packed {
13 logic is_b_zero; // Output of zero comparator
14 logic is_a_lt_b; // Output of less -than comparator
15 } ex_gcd_ss_t;
16

17 // Datapath module
18

19 module ex_gcd_GcdUnitDpath
20 (
21 input logic clk ,
22 input logic reset ,
23

24 // Data signals
25

26 input ex_gcd_req_msg_t req_msg ,
27 output ex_gcd_resp_msg_t resp_msg ,
28

29 // Control and satus signals
30

31 input ex_gcd_cs_t cs,
32 output ex_gcd_ss_t ss
33 );
34

35 localparam c_nbits = `EX_GCD_REQ_MSG_A_NBITS;
36

37 // A Mux
38

39 logic [c_nbits -1:0] b_reg_out;
40 logic [c_nbits -1:0] sub_out;
41 logic [c_nbits -1:0] a_mux_out;
42

43 vc_Mux3 #( c_nbits) a_mux
44 (
45 .sel (cs.a_mux_sel),
46 .in0 (req_msg.a),
47 .in1 (b_reg_out),
48 .in2 (sub_out),
49 .out (a_mux_out)
50 );
51

52 // A register
53

54 logic [c_nbits -1:0] a_reg_out;
55

56 vc_EnReg #( c_nbits) a_reg
57 (
58 .clk (clk),
59 .reset (reset),
60 .en (cs.a_reg_en),
61 .d (a_mux_out),
62 .q (a_reg_out)
63 );

Figure 33: Portion of GCD Datapath Unit – We use struct types to encapsulate both control and
status signals and we use a preprocessor macro from the GCD message struct to determine how to
size the datapath components.

1 // ----------------------------------------------------------------------
2 // State Outputs
3 // ----------------------------------------------------------------------
4

5 localparam a_x = 2'dx;
6 localparam a_ld = 2'd0;
7 localparam a_b = 2'd1;
8 localparam a_sub = 2'd2;
9

10 localparam b_x = 1'dx;
11 localparam b_ld = 1'd0;
12 localparam b_a = 1'd1;
13

14 task set_cs
15 (
16 input logic cs_req_rdy ,
17 input logic cs_resp_val ,
18 input logic [1:0] cs_a_mux_sel ,
19 input logic cs_a_reg_en ,
20 input logic cs_b_mux_sel ,
21 input logic cs_b_reg_en
22 );
23 begin
24 req_rdy = cs_req_rdy;
25 resp_val = cs_resp_val;
26 cs.a_reg_en = cs_a_reg_en;
27 cs.b_reg_en = cs_b_reg_en;
28 cs.a_mux_sel = cs_a_mux_sel;
29 cs.b_mux_sel = cs_b_mux_sel;
30 end
31 endtask
32

33 // Labels for Mealy transistions
34

35 logic do_swap;
36 logic do_sub;
37

38 assign do_swap = ss.is_a_lt_b;
39 assign do_sub = !ss.is_b_zero;
40

41 // Set outputs using a control signal "table"
42

43 always @(*) begin
44

45 set_cs( 0, 0, a_x , 0, b_x , 0 );
46 case ( state_reg )
47 // req resp a mux a b mux b
48 // rdy val sel en sel en
49 STATE_IDLE: set_cs( 1, 0, a_ld , 1, b_ld , 1 );
50 STATE_CALC: if ( do_swap ) set_cs( 0, 0, a_b , 1, b_a , 1 );
51 else if ( do_sub ) set_cs( 0, 0, a_sub , 1, b_x , 0 );
52 STATE_DONE: set_cs( 0, 1, a_x , 0, b_x , 0 );
53

54 endcase
55

56 end

Figure 34: Portion of GCD FSM-based Control Unit for State Outputs – We can use a task to create
a “control signal table” with one row per state or Mealy transition and one column per control signal.
Local parameters can help compactly encode various control signal values.

1 cycle src A B Areg Breg ST out sink
2

3 296: 002d00e1 > 00e1:002d(0002 0000 I ) >
4 297: # > # (00e1 002d C-) >
5 298: # > # (00b4 002d C-) >
6 299: # > # (0087 002d C-) >
7 300: # > # (005a 002d C-) >
8 301: # > # (002d 002d C-). > .
9 302: # > # (0000 002d Cs) >

10 303: # > # (002d 0000 C ) >
11 304: # > # (002d 0000 D )# > #
12 305: # > # (002d 0000 D )# > #
13 306: # > # (002d 0000 D )# > #
14 307: # > # (002d 0000 D )# > #
15 308: # > # (002d 0000 D )# > #
16 309: # > # (002d 0000 D )# > #
17 310: # > # (002d 0000 D )002d > 002d
18 311: 002200cc > 00cc:0022(002d 0000 I ) >
19 312: # > # (00cc 0022 C-) >
20 313: # > # (00aa 0022 C-) >
21 314: # > # (0088 0022 C-) >

Figure 35: Line Trace for RTL Implemen-
tation of GCD – State of A and B regis-
ters at the beginning of the cycle is shown,
along with the current state of the FSM.
I = idle, Cs = calc with swap, C- = calc with
subtract, D = done.

6.5. Exploring the GCD Implementation

As with the previous section, we have provided a simulator for evaluating the performance of the
GCD implementation. In this case, we are focusing on a single implementation with two different
input datasets. You can run the simulators and look at the average number of cycles to compute a
GCD for each input dataset like this:

% cd ${TUTROOT}/build
% make ex-gcd-sim
% ./ex-gcd-sim +input=random-a +stats
% ./ex-gcd-sim +input=random-b +stats

You can run the entire evaluation with a single command like this:

% cd ${TUTROOT}/build
% make eval-ex-gcd

H To-Do On Your Own: Optimize the GCD implementation to improve the performance on the two
given input datasets. You should be able to save a few cycles by collapsing the IDLE and CALC
states into a single state; allowing a direct transition from the CALC state back into the IDLE
state when the output is ready by the input is not valid; and allowing a direct transition from
the CALC state back into the CALC state when the output is ready and the input is valid. These
optimizations should eliminate any bubbles and improve the performance of back-to-back GCD
transactions. Another optimization approach would be to perform a swap and subtraction in the
same cycle. This will require modifying both the datapath and the control unit, but should have
a significant impact on the overall performance.

Appendix A: Constructs Allowed in Synthesizable RTL

Allowed in Explicitly
Always Allowed in Synthesizable RTL Not Allowed in
Synthesizable RTL With Limitations Synthesizable RTL

logic always1 wire, reg11

logic [N-1:0] enum2 integer, real, time, realtime
& | ^ ^~ ~ (bitwise) struct3 signed12

&& || ! casez, endcase4 ===, !==
& ~& | ~| ^ ^~ (reduction) task, endtask5 * / % ** 13

+ - function, endfunction5 #N (delay statements)
>> << >>> = (blocking assignment)6 inout14

== != > <= < <= <= (non-blocking assignment)7 initial
{} typedef8 variable initialization15

{N{}} (repeat) packed9 negedge16

?: $signed()10 casex, endcase
if else for, while, repeat, forever17

case, endcase fork, join
begin, end deassign, force, release
module, endmodule specify, endspecify
input, output nmos, pmos, cmos
assign rnmos, rpmos, rcmos
parameter tran, tranif0, tranif1
localparam rtran, rtranif0, rtranif1
genvar supply0, supply1
generate, endgenerate strong0, strong1
generate for weak0, weak1
generate if else primitive, endprimitive
generate case defparam
$clog2() unnamed port connections18

$bits() unnamed parameter passing19

named port connections all other keywords
named parameter passing all other system tasks

1 always can only be used in one of the following two constructs: always @(posedge clk) for sequential
logic, and always @(*) for combinational logic. Students are not allowed to trigger sequential blocks off of
the negative edge of the clock or create asynchronous resets, nor use explicit sensitivity lists.

2 enum can only be used with an explicit base type of logic and explicitly setting the bitwidth using the fol-
lowing syntax: typedef enum logic [$clog2(N)-1:0] { ... } type_t; where N is the number of labels
in the enum. Anonymous enums are not allowed.

3 struct can only be used with the packed qualifier (i.e., unpacked structs are not allowed) using the follow-
ing syntax: typedef struct packed { ... } type_t; Anonymous structs are not allowed.

4 casez can only be used in very specific situations to compactly implement priority encoder style hardware
structures.

5 task and function blocks must themselves contain only synthesizable RTL.

6 Blocking assignments should only be used in always @(*) blocks and are explicitly not allowed in always
@(posedge clk) blocks.

7 Non-blocking assignments should only be used in always @(posedge clk) blocks and are explicitly not
allowed in always @(*) blocks.

8 $signed() can only be used around the operands to >>>, >, >=, <, <= to ensure that these operators perform
the signed equivalents.

9 wire and reg are perfectly valid, synthesizable constructs, but logic is much cleaner. So we would like
students to avoid using wire and reg.

10 typedef should only be used in conjunction with enum and struct.

11 packed should only be used in conjunction with struct.

12 signed types can sometimes be synthesized, but we do not allow this construct in the course.

13 Multipliers can be synthesized and small multipliers can even be synthesized efficiently, but we do not
allow this construct in the course. If you need to multiply or divide by a power of two, then you can use
the left and right shift operators.

14 Ports with inout can be used to create tri-state buses, but tools often have trouble synthesizing hardware
from these kinds of models.

15 Variable initialization means assigning an initial value to a logic variable when you declare the variable.
This is not synthesizable; it is not modeling real hardware. If you need to set some state to an initial
condition, you must explicitly use the reset signal.

16 Triggering a sequential block off of the negedge of a signal is certainly synthesizable, but we will be exclu-
sively using a positive-edge-triggered flip-flop-based design style.

17 If you would like to generate hardware using loops, then you should use generate blocks.

18 In very specific, rare cases unnamed port connections might make sense, usually when there are just one or
two ports and there purpose is obvious from the context.

19 In very specific, rare cases unnamed parameter passing might make sense, usually when there are just one
or two parameters and their purpose is obvious from the context.

